Set Warnings "-notation-overridden,-parsing".
Require Export IndProp.

Definition relation (X : Type) := X -> X -> Prop.

Print le.

Check le : nat -> nat -> Prop.
Check le : relation nat.

Definition partial_function {X : Type} (R : relation X) :=
  forall x y1 y2 : X, R x y1 -> R x y2 -> y1 = y2.

Print next_nat.
Check next_nat : relation nat.

Theorem next_nat_partial_function :
  partial_function next_nat.
Proof.
  unfold partial_function.
  intros x y1 y2 H1 H2.
  inversion H1. inversion H2.
  reflexivity.
Qed.

Theorem le_not_a_partial_function :
  ~ (partial_function le).
Proof.
  unfold not. unfold partial_function. intros Hc.
  assert (0 = 1) as Nonsense. {
    apply Hc with (x := 0).
    - apply le_n.
    - apply le_S. apply le_n. }
  inversion Nonsense. Qed.

Theorem total_relation_not_a_partial_function :
  ~ (partial_function total_relation).
Proof.
  unfold not. unfold partial_function. intros Hc.
  assert (0 = 1) as Nonsense. {
    apply Hc with (x := 0).
    - apply (tr 0 0).
    - apply (tr 0 1).
  }
  inversion Nonsense. Qed.

Theorem empty_relation_is_a_partial_function :
  partial_function empty_relation.
Proof.
  unfold partial_function.
  intros x y1 y2 H1 H2.
  inversion H1. inversion H2.
  rewrite <- H6. rewrite <- H3.
  reflexivity.
Qed.

Definition reflexive {X : Type} (R: relation X) :=
  forall a : X, R a a.

Theorem le_reflexive : reflexive le.
Proof.
  unfold reflexive. intros n. apply le_n. Qed.

Definition transitive {X: Type} (R: relation X) :=
  forall a b c : X, (R a b) -> (R b c) -> (R a c).

Theorem le_trans : transitive le.
Proof.
  intros n m o Hnm Hmo.
  induction Hmo.
  - apply Hnm.
  - apply le_S. apply IHHmo. Qed.

Theorem lt_trans : transitive lt.
Proof.
  unfold lt. unfold transitive.
  intros n m o Hnm Hmo.
  apply le_S in Hnm.
  apply le_trans with (a := (S n)) (b := (S m)) (c := o).
  apply Hnm.
  apply Hmo.
Qed.

Theorem lt_trans' : transitive lt.
Proof.
  unfold lt. unfold transitive.
  intros n m o Hnm Hmo.
  induction Hmo as [| m' Hm' o].
  - inversion Hnm.
    + rewrite -> H in Hnm. rewrite -> H. apply le_S.
      apply le_n.
    + rewrite H0. apply le_S. apply Hnm.
  - apply le_S. apply o.
Qed.

Theorem lt_trans'' :
  transitive lt.
Proof.
  unfold lt. unfold transitive.
  intros n m o Hnm Hmo.
  induction o as [| o'].
  - inversion Hmo.
  - inversion Hmo.
    + rewrite <- H0. apply le_S. apply Hnm.
    + apply le_S. apply IHo'. apply H0.
Qed.

Theorem le_Sn_le : forall n m, S n <= m -> n <= m.
Proof.
  intros n m H. apply le_trans with (S n).
  - apply le_S. apply le_n.
  - apply H.
Qed.

Theorem le_S_n : forall n m,
  (S n <= S m) -> (n <= m).
Proof.
  intros n m H.
  inversion H.
  - apply le_n.
  - apply le_Sn_le. apply H1.
Qed.

Theorem le_Sn_n : forall n, ~ (S n <= n).
Proof.
  intros n. unfold not.
  intros H. induction n.
  - inversion H.
  - apply IHn. apply le_S_n. apply H.
Qed.

Definition symmetric {X : Type} (R : relation X) :=
  forall a b : X, (R a b) -> (R b a).

Theorem le_not_symmetric : ~ (symmetric le).
Proof.
  unfold not.
  unfold symmetric.
  intro H.
  assert (1 <= 0). {
    apply (H 0). apply le_S. apply le_n.
  }
  inversion H0.
Qed.

Definition antisymmetric {X : Type} (R : relation X) :=
  forall a b : X, (R a b) -> (R b a) -> a = b.

Theorem le_antisymmetric : antisymmetric le.
Proof.
  unfold antisymmetric.
  intros a b H1 H2.
  inversion H1.
  - reflexivity.
  - apply (le_trans a b a) in H1.
Admitted.

Definition equivalence {X : Type} (R : relation X) :=
  (reflexive R) /\ (symmetric R) /\ (transitive R).

Definition order { X : Type} (R : relation X) :=
  (reflexive R) /\ (antisymmetric R) /\ (transitive R).

Definition preorder {X:Type} (R: relation X) :=
  (reflexive R) /\ (transitive R).

Theorem le_order : order le.
Proof.
  unfold order. split.
  - apply le_reflexive.
  - split.
    + apply le_antisymmetric.
    + apply le_trans. Qed.

Inductive clos_refl_trans {A: Type} (R : relation A) : relation A :=
  | rt_step : forall x y, R x y -> clos_refl_trans R x y
  | rt_refl : forall x, clos_refl_trans R x x
  | rt_trans : forall x y z,
      clos_refl_trans R x y ->
      clos_refl_trans R y z ->
      clos_refl_trans R x z.

Theorem next_nat_closure_is_le : forall n m,
  (n <= m) <-> ((clos_refl_trans next_nat) n m).
Proof.
  intros n m. split.
  - intro H. induction H.
    + apply rt_refl.
    + apply rt_trans with m. apply IHle.
      apply rt_step. apply nn.
  - intro H. induction H.
    + inversion H. apply le_S. apply le_n.
    + apply le_n.
    + apply le_trans with y.
      apply IHclos_refl_trans1.
      apply IHclos_refl_trans2. Qed.

Inductive clos_refl_trans_ln {A : Type}
  (R : relation A) (x : A) : A -> Prop :=
  | rtln_refl : clos_refl_trans_ln R x x
  | rtln_trans (y z : A) :
      R x y -> clos_refl_trans_ln R y z ->
      clos_refl_trans_ln R x z.

Lemma rsc_R : forall (X : Type) (R : relation X) (x y : X),
  R x y -> clos_refl_trans_ln R x y.
Proof.
  intros X R x y H.
  apply rtln_trans with y. apply H. apply rtln_refl. Qed.

Lemma rsc_trans : forall (X:Type) (R : relation X) (x y z : X),
  clos_refl_trans_ln R x y ->
  clos_refl_trans_ln R y z ->
  clos_refl_trans_ln R x z.
Proof.
  intros X R x y z H1 H2.
  induction H1.
  - apply H2.
  - induction H1.
    +
Abort.


Require Export Basics.

Theorem plus_n_O : forall n : nat, n = n + 0.
Proof.
  induction n.
  - simpl. reflexivity.
  - simpl. rewrite <- IHn. reflexivity.
Qed.

Theorem minus_diag : forall n : nat, minus n n = 0.
Proof.
  induction n as [| n' IHn'].
  - simpl. reflexivity.
  - simpl. rewrite -> IHn'. reflexivity.
Qed.

Theorem mult_0_r : forall n : nat, n * 0 = 0.
Proof.
  induction n as [| n' IHn'].
  - simpl. reflexivity.
  - simpl. rewrite -> IHn'. reflexivity.
Qed.

Lemma plus_one : forall n : nat,
  S n = n + 1.
Proof.
  induction n.
  - simpl. reflexivity.
  - simpl. rewrite <- IHn. reflexivity.
Qed.

Theorem plus_n_Sm : forall n m : nat,
  S (n + m) = n + (S m).
Proof.
  induction n as [| n' IHn'].
  - simpl. tauto.
  - intros m. simpl. rewrite <- IHn'. reflexivity.
Qed.

Theorem plus_comm: forall n m : nat,
  n + m = m + n.
Proof.
  intros n m.
  induction n.
  - rewrite <- plus_n_O. rewrite plus_O_n. reflexivity.
  - simpl. rewrite IHn. rewrite plus_n_Sm. reflexivity.
Qed.

Theorem plus_assoc : forall n m p : nat,
  n + (m + p) = (n + m) + p.
Proof.
  intros n m p.
  induction n.
  - simpl. reflexivity.
  - simpl. rewrite IHn. reflexivity.
Qed.

Fixpoint double (n:nat) :=
  match n with
  | O => O
  | S n' => S (S (double n'))
  end.

Lemma double_plus : forall n : nat,
  double n = n + n.
Proof.
  induction n as [| n' IHn'].
  - simpl. reflexivity.
  - simpl. rewrite IHn'. rewrite <- plus_n_Sm. reflexivity.
Qed.

(*Fixpoint evenb (n : nat) : bool :=
  match n with
  | O => true
  | S O => false
  | S (S n') => evenb n'
  end.*)

Fixpoint evenb (n : nat) : bool :=
  match n with
  | O => true
  | S n' => negb (evenb n')
  end.

Theorem evenb_S : forall n : nat,
  evenb (S n) = negb (evenb n).
Proof.
  intros n.
  simpl.
  reflexivity.
Qed.

Theorem plus_rearrange : forall n m p q : nat,
  (n + m) + (p + q) = (m + n) + (p + q).
Proof.
  intros n m p q.
  assert (H: n + m = m + n).
  { rewrite -> plus_comm. reflexivity. }
  rewrite -> H. reflexivity.
Qed.


From LF Require Export Induction.

(* ================================================================= *)
(** ** Higher-Order Functions *)

(** Functions that manipulate other functions are often called
    _higher-order_ functions.  Here's a simple one: *)

Definition doit3times {X : Type} (f: X -> X) (x:X) : X :=
  f (f (f x)).

Check doit3times.

Check @doit3times.

Definition minustwo (x:nat) : nat := x - 2.

Example test_doit3times: doit3times minustwo 9 = 3.
(* minustwo (minustwo (minustwo 9) )*)
Proof. reflexivity. Qed.

(* List notation *)
Notation "x :: y" := (cons x y)
                     (at level 60, right associativity).
Notation "[ ]" := nil.
Notation "[ x ; .. ; y ]" := (cons x .. (cons y []) ..).

(** Filter function *)
Fixpoint filter {X:Type} (test: X->bool) (l:list X) : list X :=
  match l with
  | [] => []
  | h :: t => if test h then h::(filter test t)
              else filter test t
  end.

Example test_filter1: filter even [1;2;3;4] = [2;4].
Proof. reflexivity. Qed.

Definition length_is_1 {X : Type} (l : list X) : bool :=
  (List.length l) =? 1.

Example test_filter2:
    filter length_is_1
           [ [1; 2]; [3]; [4]; [5;6;7]; []; [8] ]
  = [ [3]; [4]; [8] ].
Proof. reflexivity. Qed.

(* ================================================================= *)
(** ** Anonymous Functions *)

(* Write a function [filter_lt_10] that filters all numbers less than
   10 from a given list list of numbers. *)

Compute (filter (fun xs => List.length xs =? 1)
[ [1; 2]; [3]; [4]; [5;6;7]; []; [8] ]).

(* ================================================================= *)
(** ** Map *)

(** Another handy higher-order function is called [map]. *)
Fixpoint map {X Y : Type} (f : X->Y) (l : list X) : list Y :=
  match l with
  | [] => []
  | h :: t => (f h)::(map f t)
  end.
(* Write a function [map_to_length] that maps a list of lists to list
   of their lengths *)

Definition map_to_length {X:Type} (l : list (list X)) : list nat :=
  map (fun xs => List.length xs) l.

Example test_map_to_length :
  map_to_length [ [1; 2]; [3]; [4]; [5;6;7]; []; [8] ] =
                      [2; 1; 1; 3; 0; 1].
Proof. reflexivity. Qed.

(* ================================================================= *)
(** ** Fold *)

(** An even more powerful higher-order function is called
    [fold].  This function is the inspiration for the "[reduce]"
    operation that lies at the heart of Google's map/reduce
    distributed programming framework. *)

(*
   sum[2; 3; 4; 5]
   = sum [2; 3; 4] + 5
   = sum [2;3] + 4 + 5
   .. = 2 + 3 + 4 +5

   sum[2; 3; 4; 5]
   = 2 + sum [3;4;5]
   = 2 + 3 + sum [4;5]
   = ..
   = 2 + 3 + 4 +5
*)

Fixpoint fold {X Y: Type} (f : X->Y->Y) (l : list X) (b : Y)
                         : Y :=
  match l with
  | [] => b
  | h :: t => f h (fold f t b)
  end.

Example fold_example1 :
  fold mult [1;2;3;4] 1 = 24.
Proof. reflexivity. Qed.

Check @fold.
Check fold mult.

(* ================================================================= *)
(** ** Currying and Uncurrying *)

Definition prod_curry {X Y Z : Type}
  (f : X * Y -> Z) (x : X) (y : Y) : Z := f (x, y).

(** As an exercise, define its inverse, [prod_uncurry].  Then prove
    the theorems below to show that the two are inverses. *)

Definition prod_uncurry {X Y Z : Type}
  (f : X -> Y -> Z) (p : X * Y) : Z
  (* REPLACE THIS LINE WITH ":= _your_definition_ ." *). Admitted.


From mathcomp Require Import ssreflect ssrfun ssrbool eqtype ssrnat div.
Set Implicit Arguments.
Unset Strict Implicit.
Unset Printing Implicit Defensive.

(*** Proofs by induction *)

(** * Generalizing Induction Hypothesis *)

(** The standard (non-tail-recursive) factorial function *)
Locate "`!".
Print factorial.
Print fact_rec.

Fixpoint factorial_mul (n : nat) (k : nat) : nat :=
  if n is n'.+1 then
    factorial_mul n' (n * k)
  else
    k.

Definition factorial_iter (n : nat) : nat :=
  factorial_mul n 1.

Lemma factorial_mul_correct n k :
  factorial_mul n k = n`! * k.
Proof.
elim: n k=> [|n IHn] k; first by rewrite fact0 mul1n.
by rewrite factS /= IHn mulnCA mulnA.
Qed.

Lemma factorial_iter_correct n :
  factorial_iter n = n`!.
Proof.
by rewrite /factorial_iter factorial_mul_correct muln1.
Qed.

(** * Fibonacci numbers and custom induction principles *)

(** Let's define a recursive Fibonacci function *)

(** Coq cannot figure out that we are using
    structural recursion here. It needs a hint. *)
Fail Fixpoint fib (n : nat) : nat :=
  if n is n''.+2 then fib n'' + fib n''.+1
  else n.

(** Here is the hint: name a structural subterm explicitly
    using [as]-annotation *)
Fixpoint fib (n : nat) : nat :=
  if n is (n''.+1 as n').+1 then fib n'' + fib n'
  else n.

(** Illustrate how [simpl nomatch] works *)

Section Illustrate_simpl_nomatch.
Variable n : nat.

Lemma default_behavior :
  fib n.+1 = 0.
Proof.
move=> /=.  (* fib n.+1 should not get simplified *)
Abort.

Arguments fib n : simpl nomatch.

Lemma after_simpl_nomatch :
  fib n.+2 = 0.
Proof.
move=> /=.  (* this is what we want *)
Abort.

End Illustrate_simpl_nomatch.

(** The results of the [Arguments] command does not survive
    sections so we have to repeat it here *)
Arguments fib n : simpl nomatch.

(** And here is a more efficient iterative version *)
Fixpoint fib_iter (n : nat) (f0 f1 : nat) : nat :=
  if n is n'.+1 then fib_iter n' f1 (f0 + f1)
  else f0.

Arguments fib_iter : simpl nomatch.

Lemma fib_iterS n f0 f1 :
  fib_iter n.+1 f0 f1 = fib_iter n f1 (f0 + f1).
Proof. by []. Qed.

Lemma fib_iter_sum n f0 f1 :
  fib_iter n.+2 f0 f1 =
  fib_iter n f0 f1 + fib_iter n.+1 f0 f1.
Proof.
elim: n f0 f1 => [//|n IHn] f0 f1.
by rewrite fib_iterS IHn.
Qed.

Lemma dup {A} : A -> A * A. Proof. by []. Qed.

(** This induction principle repeats, in a sense,
    the structure of the (recursive) Fibonacci function *)
Lemma nat_ind2 (P : nat -> Prop) :
  P 0 ->
  P 1 ->
  (forall n, P n -> P n.+1 -> P n.+2) ->
  forall n, P n.
Proof.
move=> p0 p1 Istep n; suffices: P n /\ P n.+1 by case.
by elim: n=> // n [/Istep pn12] /dup[/pn12].
Qed.

Lemma fib_iter_correct n :
  fib_iter n 0 1 = fib n.
Proof.
elim/nat_ind2: n=> // n IHn1 IHn2.
by rewrite fib_iter_sum IHn1 IHn2.
Qed.
(** Note: fib_iter_correct can be proven using
    suffices:
     (fib_iter n 0 1 = fib n /\ fib_iter n.+1 0 1 = fib n.+1).
 *)

(** * Another way is to provide a spec for fib_iter *)

From Coq Require Import Omega.  (* to use [omega] tactic *)
From Coq Require Import Psatz.  (* to use [lia] tactic *)

(**
- [omega] is a solver for Presburger arithmetic:
  https://en.wikipedia.org/wiki/Presburger_arithmetic

- [lia] is a solver for Linear Integer Arithmetic
*)

Lemma fib_iter_spec n f0 f1 :
  fib_iter n.+1 f0 f1 = f0 * fib n + f1 * fib n.+1.
Proof.
elim: n f0 f1=> [|n IHn] f0 f1; first by rewrite muln0 muln1.
rewrite fib_iterS IHn /=.
(** Using a bit of automation to finish off the proof *)
Fail rewrite -!plusE -!multE; omega.
by rewrite -!plusE -!multE; lia.

Restart.

(** Manual solution *)
elim: n f0 f1=> [|n IHn] f0 f1; first by rewrite muln0 muln1.
by rewrite fib_iterS IHn /= mulnDr mulnDl addnCA.
Qed.

Lemma fib_iter_correct' n :
  fib_iter n 0 1 = fib n.
Proof.
by case: n=> // n; rewrite fib_iter_spec mul1n.
Qed.

(** * Yet another solutiton *)

(* due to D.A. Turner, see his "Total Functional Programming" (2004) *)
Lemma fib_iter_spec' n p :
  fib_iter n (fib p) (fib p.+1) = fib (p + n).
Proof.
elim: n p=> [|n IHn] p; first by rewrite addn0.
by rewrite fib_iterS IHn addnS.
Qed.

Lemma fib_iter_correct'' n :
  fib_iter n 0 1 = fib n.
Proof.
Fail apply: fib_iter_spec'.
by apply: (fib_iter_spec' n 0).
(* Alternative (longer, but more explicit) solution: *)
Restart.
suffices: (fib_iter n (fib 0) (fib 1) = fib n) by [].
by apply: fib_iter_spec'.
Qed.

(** * Complete induction *)

(** It's also called:
    - strong induction;
    - well-founded induction;
    - course-of-values induction
 *)

Lemma lt_wf_ind (P : nat -> Prop) :
  (forall m, (forall k : nat, (k < m) -> P k) -> P m) ->
  forall n, P n.
Proof.
  (* exercise! *)
Admitted.

(** In SSReflect/Mathcomp one does not use
    a custom principle like above
    directly, but rather generates it on the fly
    using [leqnn] lemma.
 *)

Lemma fib_iter_correct''' n :
  fib_iter n 0 1 = fib n.
Proof.
move: (leqnn n).
move: {-2}n.
move: n.
elim.
case.
done.
by case.
move=> n IHn.
(* ^ the proof steps above correspond to one line below
   marked <- *)

Restart.

elim: n {-2}n (leqnn n)=> [[]//|n IHn].  (* <- *)
case=> //; case=> // n0.
rewrite fib_iter_sum.
move=> /dup[/ltnW/IHn-> ].
by rewrite ltnS=> /IHn->.
Qed.

(*** Lists *)

From mathcomp Require Import ssrnat ssrbool eqtype seq path.
(** Note that we added [seq] and [path] modules to imports *)

(** [seq] is a Mathcomp's notation for [list] data type *)
Print seq.
Print list.

(**
   Inductive list (A : Type) : Type :=
   | nil : seq A
   | cons : A -> seq A -> seq A
*)

(** A simple example *)
Compute [:: 1; 2; 3] ++ [::].

(** List concatenation *)
Locate "++".
Print cat.

(** * Structural Induction for Lists *)

Section StructuralInduction.

Variable T : Type.

Implicit Types xs ys zs : seq T.

Lemma catA xs ys zs :
  xs ++ (ys ++ zs) = (xs ++ ys) ++ zs.
Proof.

Check list_ind :
  forall (A : Type) (P : seq A -> Prop),
    P [::] ->
    (forall (a : A) (l : seq A), P l -> P (a :: l)) ->
    forall l : seq A, P l.

by elim: xs => //= x xs' ->.
Qed.

End StructuralInduction.

(** * Classical example: list reversal function *)

(** The standard implementation is tail recursive *)
Print rev.
Print catrev.

Fixpoint rev_rec {A : Type} (xs : seq A) : seq A :=
  if xs is (x::xs') then
    rev_rec xs' ++ [:: x]
  else xs.

Lemma rev_rec_inv A :
  involutive (@rev A).
Proof.
Admitted.

Lemma rev_correct A (xs : seq A):
  rev xs = rev_rec xs.
Proof.
Admitted.


(* QUIZ *)
(** Recall the types of [cons] and [nil]:
       nil : forall X : Type, list X
       cons : forall X : Type, X -> list X -> list X
    What is the type of [cons bool true (cons nat 3 (nil nat))]?

    (1) [nat -> list nat -> list nat]

    (2) [forall (X:Type), X -> list X -> list X]

    (3) [list nat]

    (4) [list bool]

    (5) No type can be assigned.
*)

Inductive list (X:Type) : Type :=
  | nil : list X
  | cons : X -> list X -> list X.

Fail Check (cons bool true (cons nat 3 (nil nat))).

(* The command has indeed failed with message:
The term "cons nat 3 (nil nat)" has type "list nat"
while it is expected to have type "list bool". *)

(* /QUIZ *)

(* QUIZ *)
(** Recall the definition of [repeat]: *)
      Fixpoint repeat (X : Type) (x : X) (count : nat) : list X :=
        match count with
        | 0 => nil X
        | S count' => cons X x (repeat X x count')
        end.
(*  What is the type of [repeat?

    (1) [nat -> nat -> list nat]

    (2) [X -> nat -> list X]

    (3) [forall (X Y:Type), X -> nat -> list Y]

    (4) [forall (X:Type), X -> nat -> list X]

    (5) No type can be assigned.

*)

Check repeat.
(* repeat
     : forall X : Type, X -> nat -> list X *)
(* /QUIZ *)

(* QUIZ *)
(** What is the type of [repeat nat 1 2]?

    (1) [list nat]

    (2) [forall (X:Type), X -> nat -> list X]

    (3) [list bool]

    (4) No type can be assigned.

*)
Check (repeat nat 1 2).
Arguments nil {X}.
Arguments cons {X} _ _.
Notation "x :: l" := (cons x l)
                     (at level 60, right associativity).
Notation "[ ]" := nil.
Notation "[ x ; .. ; y ]" := (cons x .. (cons y []) ..).

Check (repeat nat 1 2).
(* repeat nat 1 2
     : list nat *)

(* /QUIZ *)

(* QUIZ *)

(** Which type does Coq assign to the following expression?
    [1;2;3]
    (1) [list nat]

    (2) [list bool]

    (3) [bool]

    (4) No type can be assigned
*)

Check [1;2;3].
(* [1; 2; 3]
     : list nat *)
(* /QUIZ *)

(* QUIZ *)
(** Which type does Coq assign to the following expression?
    [3 + 4] ++ nil
    (1) [list nat]

    (2) [list bool]

    (3) [bool]

    (4) No type can be assigned
*)

Fixpoint app {X : Type} (l1 l2 : list X)
             : (list X) :=
  match l1 with
  | nil      => l2
  | cons h t => cons h (app t l2)
  end.

Notation "x ++ y" := (app x y)
                     (at level 60, right associativity).

Check @app.
(* app
     : forall X : Type, list X -> list X -> list X *)

Check ([3 + 4] ++ nil).
(* [3 + 4] ++ [ ]
     : list nat *)
(* /QUIZ *)

(* QUIZ *)
(** What about this one?
    [andb true false ++ nil]
    (1) [list nat]

    (2) [list bool]

    (3) [bool]

    (4) No type can be assigned
*)

Fail Check [andb true false ++ nil].
(* The command has indeed failed with message:
The term "(true && false)%bool" has type "bool"
while it is expected to have type "list ?X0" *)

(* /QUIZ *)

(* QUIZ *)
(** What about this one?
    [1; nil]
    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
*)

Fail Check [1;nil].
(* The command has indeed failed with message:
The term "[[ ]]" has type "list (list ?X)" while it is expected to have type
 "list nat". *)

(* /QUIZ *)

(* QUIZ *)
(** What about this one?

        [[1]; nil]

    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
 *)

Check [[1]; nil].
(* [[1]; [ ]]
     : list (list nat) *)

(* /QUIZ *)

(* QUIZ *)
(** And what about this one?

         [1] :: [nil]

    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
*)

Check [1] :: [nil].
(* [[1]; [ ]]
     : list (list nat) *)

(* /QUIZ *)

(* QUIZ *)
(** What about this one?

        @nil bool

    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
*)

Check (@nil bool).
(* [ ]
     : list bool *)

(* /QUIZ *)

(* QUIZ *)
(** What about this one?

        nil bool

    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
*)

Fail Check (nil bool).

(* The command has indeed failed with message:
Illegal application (Non-functional construction):
The expression "[ ]" of type "list ?X" cannot be applied to the term
 "bool" : "Set" *)

(* /QUIZ *)

(* QUIZ *)
(** What about this one?
   [@nil 3]
    (1) [list nat]

    (2) [list (list nat)]

    (3) [list bool]

    (4) No type can be assigned
*)
Check nil.
Check @nil.
Check (@nil nat).

Fail Check (@nil 3).
(* The command has indeed failed with message:
The term "3" has type "nat" while it is expected to have type "Type". *)

(* /QUIZ *)

(* QUIZ *)
(** Recall the definition of [map]: *)
      Fixpoint map {X Y:Type} (f:X->Y) (l:list X)
                   : (list Y) :=
        match l with
        | []     => []
        | h :: t => (f h) :: (map f t)
        end.
(*    What is the type of [map]?

    (1) [forall X Y : Type, X -> Y -> list X -> list Y]

    (2) [X -> Y -> list X -> list Y]

    (3) [forall X Y : Type, (X -> Y) -> list X -> list Y]

    (4) [forall X : Type, (X -> X) -> list X -> list X]

*)

Check @map.
(* map
     : forall X Y : Type, (X -> Y) -> list X -> list Y *)

(* /QUIZ *)

(* QUIZ *)
(** Recall that [evenb] has type [nat -> bool].

    What is the type of [map evenb]?

    (1) [forall X Y : Type, (X -> Y) -> list X -> list Y]

    (2) [list nat -> list bool]

    (3) [list nat -> list Y]

    (4) [forall Y : Type, list nat -> list Y]

*)

Fixpoint evenb (n:nat) : bool :=
  match n with
  | O        => true
  | S O      => false
  | S (S n') => evenb n'
  end.

Check (map evenb).

(* map evenb
     : list nat -> list bool *)

(* /QUIZ *)

(* QUIZ *)
(** Here is the definition of [fold] again: *)
Fixpoint fold {X Y:Type} (f: X->Y->Y) (l:list X) (b:Y) : Y :=
  match l with
  | nil => b
  | h :: t => f h (fold f t b)
  end.

(*    What is the type of [fold]?

    (1) [forall X Y : Type, (X -> Y -> Y) -> list X -> Y -> Y]

    (2) [X -> Y -> (X -> Y -> Y) -> list X -> Y -> Y]

    (3) [forall X Y : Type, X -> Y -> Y -> list X -> Y -> Y]

    (4) [X -> Y->  X -> Y -> Y -> list X -> Y -> Y]

*)

Check @fold.

(* fold
     : forall X Y : Type, (X -> Y -> Y) -> list X -> Y -> Y *)

(* /QUIZ *)

(* QUIZ *)
(** What is the type of [fold plus]?

    (1) [forall X Y : Type, list X -> Y -> Y]

    (2) [nat -> nat -> list nat -> nat -> nat]

    (3) [forall Y : Type, list nat -> Y -> nat]

    (4) [list nat -> nat -> nat]

    (5) [forall X Y : Type, list nat -> nat -> nat]

*)

Fixpoint plus (n : nat) (m : nat) : nat :=
  match n with
    | O => m
    | S n' => S (plus n' m)
  end.

Check (fold plus).
(* fold plus
     : list nat -> nat -> nat *)

(* /QUIZ *)

(* QUIZ *)
(** What does [fold plus [1;2;3;4] 0] simplify to?

   (1) [[1;2;3;4]]

   (2) [0]

   (3) [10]

   (4) [[3;7;0]]

*)

Compute (fold plus [1;2;3;4] 0).
(* 10 : nat *)
(* /QUIZ *)


Inductive ev : nat -> Prop :=
| ev_0 : ev 0
| ev_SS : forall n : nat, ev n -> ev (S (S n)).

Theorem ev_4 : ev 4.
Proof.
apply ev_SS.
apply ev_SS.
apply ev_0.
Qed.

Theorem ev_4' : ev 4.
Proof.
apply (ev_SS 2 (ev_SS 0 ev_0)). Qed.

Theorem ev_plus4 : forall n, ev n -> ev (4 + n).
Proof.
intros n.
simpl.
intros Hn.
apply ev_SS.
apply ev_SS.
apply Hn.
Qed.

Fixpoint evenb (n:nat) : bool :=
  match n with
  | O        => true
  | S O      => false
  | S (S n') => evenb n'
  end.

Definition double n := n * 2.

Theorem ev_double : forall n,
  ev (double n).
Proof.
induction n  as [| n'].
apply ev_0.
simpl.
apply ev_SS.
apply IHn'.
Qed.

Theorem evenb_minus2: forall n,
  evenb n = true -> evenb (pred (pred n)) = true.
Proof.
intros [| [| n ]].
reflexivity.
intros H.
inversion H.
simpl.
intros H.
apply H.
Qed.

Theorem ev_minus2: forall n,
  ev n -> ev (pred (pred n)).
Proof.
intros n H.
inversion H.
simpl.
apply ev_0.
simpl.
apply H0.
Qed.

Theorem evSS_ev : forall n,
  ev (S (S n)) -> ev n.
Proof.
intros n H.
inversion H.
apply H1.
Qed.

Theorem one_not_even : ~ ev 1.
Proof.
intros H.
inversion H.
Qed.

Theorem SSSSev__even : forall n, ev (S (S (S (S n)))) -> ev n.
Proof.
intros n H.
inversion H.
inversion H1.
apply H3.
Qed.

Theorem even5_nonsense :
  ev 5 -> 2 + 2 = 9.
Proof.
intros absurd.
inversion absurd.
inversion H0.
inversion H2.
Qed.

Lemma ev_even : forall n,
  ev n -> exists k, n = double k.
Proof.
  intros n E.
  induction E as [|n' E' IH].
    exists 0. reflexivity.
    destruct IH as [k' Hk'].
    rewrite Hk'. exists(S k'). reflexivity.
Qed.

Theorem ev_even_iff : forall n,
  ev n <-> exists k, n = double k.
Proof.
intros.
split.
apply ev_even.
intros [k Hk].
rewrite Hk.
apply ev_double.
Qed.

Theorem ev_sum : forall n m, ev n -> ev m -> ev (n + m).
Proof.
intros.
induction H as [| h].
simpl.
apply H0.
simpl.
apply ev_SS.
apply IHev.
Qed.

Inductive ev' : nat -> Prop :=
| ev'_0 : ev' 0
| ev'_2 : ev' 2
| ev'_sum : forall n m, ev' n -> ev' m -> ev' (n + m).

Theorem ev'_ev : forall n, ev' n <-> ev n.
Proof.
intros n.
split.
Abort.

Theorem ev_ev__ev : forall n m,
  ev (n+m) -> ev n -> ev m.
Proof.
intros n m E En.
generalize dependent E.
induction En.
simpl.
intros.
apply E.
simpl.
intros.
inversion E.
apply IHEn in H0.
apply H0.
Qed.


Set Warnings "-notation-overridden,-parsing".
From LF Require Export ProofObjects.

Check nat_ind.

Theorem mult_0_r' : forall n:nat,
  n * 0 = 0.
Proof.
  apply nat_ind.
  - reflexivity.
  - simpl. intros n' IHn'. rewrite -> IHn'.
    reflexivity. Qed.

(*standard, optional (plus_one_r')*)
Theorem plus_one_r' : forall n:nat,
  n + 1 = S n.
Proof.
  apply nat_ind.
  - reflexivity.
  - intros n H. simpl. rewrite -> H. reflexivity. Qed.
(*/standard, optional (plus_one_r')*)

Inductive yesno : Type :=
  | yes
  | no.

Check yesno_ind.

(*standard, optional (rgb)*)
(*forall P : rgb -> Prop, P red -> P green -> P blue -> forall y : rgb, P y*)
Inductive rgb : Type :=
  | red
  | green
  | blue.
Check rgb_ind.
(*/standard, optional (rgb)*)

Inductive natlist : Type :=
  | nnil
  | ncons (n : nat) (l : natlist).

Check natlist_ind.

(*standard, optional (natlist1)*)
(*forall P : natlist1 -> Prop,
       P nnil1 ->
       (forall l : natlist1, P l -> forall n : nat, P (nsnoc1 l n)) ->
       forall n : natlist1, P n*)
Inductive natlist1 : Type :=
  | nnil1
  | nsnoc1 (l : natlist1) (n : nat).

Check natlist1_ind.
(*/standard, optional (natlist1)*)

(*standard, optional (byntree_ind)*)
(*forall P : byntree -> Prop,
       P bempty ->
       (forall yn : yesno, P (bleaf yn)) ->
       (forall (yn : yesno) (t1 : byntree),
        P t1 -> forall t2 : byntree, P t2 -> P (nbranch yn t1 t2)) ->
       forall b : byntree, P b*)
Inductive byntree : Type :=
 | bempty
 | bleaf (yn : yesno)
 | nbranch (yn : yesno) (t1 t2 : byntree).

Check byntree_ind.
(*/standard, optional (byntree_ind)*)

(*standard, optional (ex_set)*)
Inductive ExSet : Type :=
  | con1 : bool -> ExSet
  | con2 : nat -> ExSet -> ExSet.
(*/standard, optional (ex_set)*)

(*standard, optional (tree)*)
(*forall (X : Type) (P : tree X -> Prop),
       (forall x : X, P (leaf X x)) ->
       (forall t1 : tree X, P t1 -> forall t2 : tree X, P t2 -> P (node X t1 t2)) ->
       forall t : tree X, P t*)
Inductive tree (X:Type) : Type :=
  | leaf (x : X)
  | node (t1 t2 : tree X).
Check tree_ind.
(*/standard, optional (tree)*)

(*standard, optional (mytype)*)
Inductive mytype (X: Type) : Type :=
  | constr1 (x : X)
  | constr2 (n : nat)
  | constr3 (m : mytype X) (n : nat).
Check mytype_ind.
(*/standard, optional (mytype)*)

(*standard, optional (foo)*)
Inductive foo (X:Type) (Y:Type) :=
  | bar (x : X)
  | baz (y : Y)
  | quux (f1 : nat -> foo X Y).

Check foo_ind.
(*/standard, optional (foo)*)

(*standard, optional (foo')*)
Inductive foo' (X:Type) : Type :=
  | C1 (l : list X) (f : foo' X)
  | C2.

Check foo'_ind.
(*/standard, optional (foo')*)

Definition P_m0r (n:nat) : Prop :=
  n * 0 = 0.

Definition P_m0r' : nat->Prop :=
  fun n => n * 0 = 0.

Theorem mult_0_r'' : forall n:nat,
  P_m0r n.
Proof.
  apply nat_ind.
  - unfold P_m0r. simpl. reflexivity.
  - intros n IHn.
    unfold P_m0r in IHn. unfold P_m0r. simpl. apply IHn. Qed.

Theorem plus_assoc' : forall n m p : nat,
  n + (m + p) = (n + m) + p.
Proof.
  intros n m p.
  induction n as [| n'].
  - reflexivity.
  -

    simpl. rewrite -> IHn'. reflexivity. Qed.

Theorem plus_comm' : forall n m : nat,
  n + m = m + n.
Proof.
  induction n as [| n'].
  - intros m. rewrite <- plus_n_O. reflexivity.
  - intros m. simpl. rewrite -> IHn'.
    rewrite <- plus_n_Sm. reflexivity. Qed.

Theorem plus_comm'' : forall n m : nat,
  n + m = m + n.
Proof.
  induction m as [| m'].
  - simpl. rewrite <- plus_n_O. reflexivity.
  - simpl. rewrite <- IHm'.
    rewrite <- plus_n_Sm. reflexivity. Qed.

Inductive even : nat -> Prop :=
| ev_0 : even 0
| ev_SS : forall n : nat, even n -> even (S (S n)).

Check even_ind.

Inductive even' : nat -> Prop :=
| even'_0 : even' 0
| even'_2 : even' 2
| even'_sum n m (Hn : even' n) (Hm : even' m) : even' (n + m).

Theorem ev_ev' : forall n, even n -> even' n.
Proof.
  apply even_ind.
  -
    apply even'_0.
  -
    intros m Hm IH.
    apply (even'_sum 2 m).
    + apply even'_2.
    + apply IH.
Qed.

Inductive le (n:nat) : nat -> Prop :=
  | le_n : le n n
  | le_S m (H : le n m) : le n (S m).

Notation "m <= n" := (le m n).

Check le_ind.


(** * Logic: Logic in Coq *)

Require Export Tactics.
Require Export Basics.

(** In previous chapters, we have seen many examples of factual
    claims (_propositions_) and ways of presenting evidence of their
    truth (_proofs_).  In particular, we have worked extensively with
    _equality propositions_ of the form [e1 = e2], with
    implications ([P -> Q]), and with quantified propositions ([forall
    x, P]).  In this chapter, we will see how Coq can be used to carry
    out other familiar forms of logical reasoning.

    Before diving into details, let's talk a bit about the status of
    mathematical statements in Coq.  Recall that Coq is a _typed_
    language, which means that every sensible expression in its world
    has an associated type.  Logical claims are no exception: any
    statement we might try to prove in Coq has a type, namely [Prop],
    the type of _propositions_.  We can see this with the [Check]
    command: *)

Check 3 = 3.
(* ===> Prop *)

Check forall n m : nat, n + m = m + n.
(* ===> Prop *)

(** Note that all well-formed propositions have type [Prop] in Coq,
    regardless of whether they are true or not. Simply _being_ a
    proposition is one thing; being _provable_ is something else! *)

Check forall n : nat, n = 2.
(* ===> Prop *)

Check 3 = 4.
(* ===> Prop *)

(** Indeed, propositions don't just have types: they are _first-class
    objects_ that can be manipulated in the same ways as the other
    entities in Coq's world.  So far, we've seen one primary place
    that propositions can appear: in [Theorem] (and [Lemma] and
    [Example]) declarations. *)

Theorem plus_2_2_is_4 :
  2 + 2 = 4.
Proof. reflexivity.  Qed.

(** But propositions can be used in many other ways.  For example, we
    can give a name to a proposition using a [Definition], just as we
    have given names to expressions of other sorts. *)

Definition plus_fact : Prop  :=  2 + 2 = 4.
Check plus_fact.
(* ===> plus_fact : Prop *)

(** We can later use this name in any situation where a proposition is
    expected -- for example, as the claim in a [Theorem] declaration. *)

Theorem plus_fact_is_true :
  plus_fact.
Proof. reflexivity.  Qed.

(** We can also write _parameterized_ propositions -- that is,
    functions that take arguments of some type and return a
    proposition. For instance, the following function takes a number
    and returns a proposition asserting that this number is equal to
    three: *)

Definition is_three (n : nat) : Prop :=
  n = 3.
Check is_three.
(* ===> nat -> Prop *)

(** In Coq, functions that return propositions are said to define
    _properties_ of their arguments.  For instance, here's a
    polymorphic property defining the familiar notion of an _injective
    function_. *)

Definition injective {A B} (f : A -> B) :=
  forall x y : A, f x = f y -> x = y.

Lemma succ_inj : injective S.
Proof.
  intros n m H. inversion H. reflexivity.
Qed.

(** The equality operator [=] that we have been using so far is also
    just a function that returns a [Prop]. The expression [n = m] is
    just syntactic sugar for [eq n m], defined using Coq's [Notation]
    mechanism. Because [=] can be used with elements of any type, it
    is also polymorphic: *)

Check @eq.
(* ===> forall A : Type, A -> A -> Prop *)

(** (Notice that we wrote [@eq] instead of [eq]: The type argument [A]
    to [eq] is declared as implicit, so we need to turn off implicit
    arguments to see the full type of [eq].) *)

(* ################################################################# *)
(** * Logical Connectives *)

(* ================================================================= *)
(** ** Conjunction *)

(** The _conjunction_ or _logical and_ of propositions [A] and [B] is
    written [A /\ B], denoting the claim that both [A] and [B] are
    true. *)

Example and_example : 3 + 4 = 7 /\ 2 * 2 = 4.

(** To prove a conjunction, use the [split] tactic.  Its effect is to
    generate two subgoals, one for each part of the statement: *)

Proof.
  split.
  - (* 3 + 4 = 7 *) reflexivity.
  - (* 2 + 2 = 4 *) reflexivity.
Qed.

(** More generally, the following principle works for any two
    propositions [A] and [B]: *)

Lemma and_intro : forall A B : Prop, A -> B -> A /\ B.
Proof.
  intros A B HA HB. split.
  - apply HA.
  - apply HB.
Qed.

(** A logical statement with multiple arrows is just a theorem that
    has several hypotheses.  Here, [and_intro] says that, for any
    propositions [A] and [B], if we assume that [A] is true and we
    assume that [B] is true, then [A /\ B] is also true.

    Since applying a theorem with hypotheses to some goal has the
    effect of generating as many subgoals as there are hypotheses for
    that theorem, we can, apply [and_intro] to achieve the same effect
    as [split]. *)

Example and_example' : 3 + 4 = 7 /\ 2 * 2 = 4.
Proof.
  apply and_intro.
  - (* 3 + 4 = 7 *) reflexivity.
  - (* 2 + 2 = 4 *) reflexivity.
Qed.

Lemma plus_O_lem : forall n m, n + m = 0 -> n = 0.
Proof.
  induction n.
  - intros. reflexivity.
    - intros. simpl in H. inversion H.
Qed.
    (** **** Exercise: 2 stars (and_exercise)  *)
Example and_exercise :
  forall n m : nat, n + m = 0 -> n = 0 /\ m = 0.
Proof.
  intros. split.
  - apply (plus_O_lem _ _ H).
  - rewrite plus_comm in H. apply (plus_O_lem _ _ H).
Qed.

(** So much for proving conjunctive statements.  To go in the other
    direction -- i.e., to _use_ a conjunctive hypothesis to prove
    something else -- we employ the [destruct] tactic.

    If the proof context contains a hypothesis [H] of the form [A /\
    B], writing [destruct H as [HA HB]] will remove [H] from the
    context and add two new hypotheses: [HA], stating that [A] is
    true, and [HB], stating that [B] is true.  For instance: *)

Lemma and_example2 :
  forall n m : nat, n = 0 /\ m = 0 -> n + m = 0.
Proof.
  intros n m H.
  destruct H as [Hn Hm].
  rewrite Hn. rewrite Hm.
  reflexivity.
Qed.

(** As usual, we can also destruct [H] when we introduce it instead of
    introducing and then destructing it: *)

Lemma and_example2' :
  forall n m : nat, n = 0 /\ m = 0 -> n + m = 0.
Proof.
  intros n m [Hn Hm].
  rewrite Hn. rewrite Hm.
  reflexivity.
Qed.

(** You may wonder why we bothered packing the two hypotheses [n = 0]
    and [m = 0] into a single conjunction, since we could have also
    stated the theorem with two separate premises: *)

Lemma and_example2'' :
  forall n m : nat, n = 0 -> m = 0 -> n + m = 0.
Proof.
  intros n m Hn Hm.
  rewrite Hn. rewrite Hm.
  reflexivity.
Qed.

(** In this case, there is not much difference between the two
    theorems.  But it is often necessary to explicitly decompose
    conjunctions that arise from intermediate steps in proofs,
    especially in bigger developments.  Here's a simplified
    example: *)

Lemma and_example3 :
  forall n m : nat, n + m = 0 -> n * m = 0.
Proof.
  intros n m H.
  assert (H' : n = 0 /\ m = 0).
  { apply and_exercise. apply H. }
  destruct H' as [Hn Hm].
  rewrite Hn. reflexivity.
Qed.

(** Another common situation with conjunctions is that we know [A /\
    B] but in some context we need just [A] (or just [B]).  The
    following lemmas are useful in such cases: *)

Lemma proj1 : forall P Q : Prop,
  P /\ Q -> P.
Proof.
  intros P Q [HP HQ].
  apply HP.  Qed.

(** **** Exercise: 1 star, optional (proj2)  *)
Lemma proj2 : forall P Q : Prop,
  P /\ Q -> Q.
Proof.
  intros. destruct H. assumption.
Qed.

(** [] *)

(** Finally, we sometimes need to rearrange the order of conjunctions
    and/or the grouping of conjuncts in multi-way conjunctions.  The
    following commutativity and associativity theorems come in handy
    in such cases. *)

Theorem and_commut : forall P Q : Prop,
  P /\ Q -> Q /\ P.
Proof.
  (* WORKED IN CLASS *)
  intros P Q [HP HQ].
  split.
    - (* left *) apply HQ.
    - (* right *) apply HP.  Qed.

(** **** Exercise: 2 stars (and_assoc)  *)
(** (In the following proof of associativity, notice how the _nested_
    intro pattern breaks the hypothesis [H : P /\ (Q /\ R)] down into
    [HP : P], [HQ : Q], and [HR : R].  Finish the proof from
    there.) *)

Theorem and_assoc : forall P Q R : Prop,
  P /\ (Q /\ R) -> (P /\ Q) /\ R.
Proof.
  intros P Q R [HP [HQ HR]].
  - apply and_intro. apply (and_intro _ _ HP HQ). assumption.
Qed.

(** [] *)

(** By the way, the infix notation [/\] is actually just syntactic
    sugar for [and A B].  That is, [and] is a Coq operator that takes
    two propositions as arguments and yields a proposition. *)

Check and.
(* ===> and : Prop -> Prop -> Prop *)

(* ================================================================= *)
(** ** Disjunction *)

(** Another important connective is the _disjunction_, or _logical or_
    of two propositions: [A \/ B] is true when either [A] or [B]
    is.  (Alternatively, we can write [or A B], where [or : Prop ->
    Prop -> Prop].)

    To use a disjunctive hypothesis in a proof, we proceed by case
    analysis, which, as for [nat] or other data types, can be done
    with [destruct] or [intros].  Here is an example: *)

Lemma or_example :
  forall n m : nat, n = 0 \/ m = 0 -> n * m = 0.
Proof.
  (* This pattern implicitly does case analysis on
     [n = 0 \/ m = 0] *)
  intros n m [Hn | Hm].
  - (* Here, [n = 0] *)
    rewrite Hn. reflexivity.
  - (* Here, [m = 0] *)
    rewrite Hm. rewrite <- mult_n_O.
    reflexivity.
Qed.

(** We can see in this example that, when we perform case analysis on
    a disjunction [A \/ B], we must satisfy two proof obligations,
    each showing that the conclusion holds under a different
    assumption -- [A] in the first subgoal and [B] in the second.
    Note that the case analysis pattern ([Hn | Hm]) allows us to name
    the hypothesis that is generated in each subgoal.

    Conversely, to show that a disjunction holds, we need to show that
    one of its sides does. This is done via two tactics, [left] and
    [right].  As their names imply, the first one requires proving the
    left side of the disjunction, while the second requires proving
    its right side.  Here is a trivial use... *)

Lemma or_intro : forall A B : Prop, A -> A \/ B.
Proof.
  intros A B HA.
  left.
  apply HA.
Qed.

(** ... and a slightly more interesting example requiring the use of
    both [left] and [right]: *)

Lemma zero_or_succ :
  forall n : nat, n = 0 \/ n = S (pred n).
Proof.
  intros [|n].
  - left. reflexivity.
  - right. reflexivity.
Qed.

(** **** Exercise: 1 star (mult_eq_0)  *)
Lemma mult_eq_0 :
  forall n m, n * m = 0 -> n = 0 \/ m = 0.
Proof.
  intros n m. destruct n.
  - intros. left. reflexivity.
  - intros. simpl in H. right. pose proof (and_exercise _ _ H) as H0.
    destruct H0. assumption.
Qed.

(** [] *)

Lemma or_intro_r : forall A B : Prop, B -> A \/ B.
Proof.
  intros A B HB.
  right.
  apply HB.
Qed.

(** **** Exercise: 1 star (or_commut)  *)
Theorem or_commut : forall P Q : Prop,
  P \/ Q  -> Q \/ P.
Proof.
  intros P Q [].
  - intros. apply (or_intro_r _ _ H).
  - intros. apply (or_intro _ _ H).
Qed.

(** [] *)

(* ================================================================= *)
(** ** Falsehood and Negation *)

(** So far, we have mostly been concerned with proving that certain
    things are _true_ -- addition is commutative, appending lists is
    associative, etc.  Of course, we may also be interested in
    _negative_ results, showing that certain propositions are _not_
    true. In Coq, such negative statements are expressed with the
    negation operator [~].

    To see how negation works, recall the discussion of the _principle
    of explosion_ from the [Tactics] chapter; it asserts that, if we
    assume a contradiction, then any other proposition can be derived.
    Following this intuition, we could define [~ P] ("not [P]") as
    [forall Q, P -> Q].  Coq actually makes a slightly different
    choice, defining [~ P] as [P -> False], where [False] is a
    _particular_ contradictory proposition defined in the standard
    library. *)

Module MyNot.

Definition not (P:Prop) := P -> False.

Notation "~ x" := (not x) : type_scope.

Check not.
(* ===> Prop -> Prop *)

End MyNot.

(** Since [False] is a contradictory proposition, the principle of
    explosion also applies to it. If we get [False] into the proof
    context, we can [destruct] it to complete any goal: *)

Theorem ex_falso_quodlibet : forall (P:Prop),
  False -> P.
Proof.
  (* WORKED IN CLASS *)
  intros P contra.
  destruct contra.  Qed.

(** The Latin _ex falso quodlibet_ means, literally, "from falsehood
    follows whatever you like"; this is another common name for the
    principle of explosion. *)

(** **** Exercise: 2 stars, optional (not_implies_our_not)  *)
(** Show that Coq's definition of negation implies the intuitive one
    mentioned above: *)

Fact not_implies_our_not : forall (P:Prop),
  ~ P -> (forall (Q:Prop), P -> Q).
Proof.
  intros. unfold not in H. apply (ex_falso_quodlibet _ (H H0)).
Qed.

(** [] *)

(** This is how we use [not] to state that [0] and [1] are different
    elements of [nat]: *)

Theorem zero_not_one : ~(0 = 1).
Proof.
  intros contra. inversion contra.
Qed.

(** Such inequality statements are frequent enough to warrant a
    special notation, [x <> y]: *)

Check (0 <> 1).
(* ===> Prop *)

Theorem zero_not_one' : 0 <> 1.
Proof.
  intros H. inversion H.
Qed.

(** It takes a little practice to get used to working with negation in
    Coq.  Even though you can see perfectly well why a statement
    involving negation is true, it can be a little tricky at first to
    get things into the right configuration so that Coq can understand
    it!  Here are proofs of a few familiar facts to get you warmed
    up. *)

Theorem not_False :
  ~ False.
Proof.
  unfold not. intros H. destruct H. Qed.

Theorem contradiction_implies_anything : forall P Q : Prop,
  (P /\ ~P) -> Q.
Proof.
  (* WORKED IN CLASS *)
  intros P Q [HP HNA]. unfold not in HNA.
  apply HNA in HP. destruct HP.  Qed.

Theorem double_neg : forall P : Prop,
  P -> ~~P.
Proof.
  (* WORKED IN CLASS *)
  intros P H. unfold not. intros G. apply G. apply H.  Qed.

(** **** Exercise: 2 stars, advanced, recommended (double_neg_inf)  *)
(** Write an informal proof of [double_neg]:

   _Theorem_: [P] implies [~~P], for any proposition [P].

   _Proof_:
   Since ~P expands to (P -> _|_) what we need to prove is
   H:P -> (G:P -> _|_) -> _|_
   Since we assume H which is a proof of P, we can apply G to obtain _|_
   and we're done.
   []
*)

(** **** Exercise: 2 stars, recommended (contrapositive)  *)
Theorem contrapositive : forall P Q : Prop,
  (P -> Q) -> (~Q -> ~P).
Proof.
  intros. unfold not in *. intros. apply (H0 (H H1)).
Qed.

(** [] *)

(** **** Exercise: 1 star (not_both_true_and_false)  *)
Theorem not_both_true_and_false : forall P : Prop,
  ~ (P /\ ~P).
Proof.
  unfold not. intros P []. intros. apply (H0 H).
Qed.

(** [] *)

(** **** Exercise: 1 star, advanced (informal_not_PNP)  *)
(** Write an informal proof (in English) of the proposition [forall P
    : Prop, ~(P /\ ~P)]. *)

(* We have to prove (P /\ (P -> _|_)) -> _|_.
   We can assume H:(P /\ (P -> _|_)). Using and elimination we can also
   assume H1:P and H2:P -> _|_. Applying H with H1 we get _|_ and we're done.
 *)
(** [] *)

(** Similarly, since inequality involves a negation, it requires a
    little practice to be able to work with it fluently.  Here is one
    useful trick.  If you are trying to prove a goal that is
    nonsensical (e.g., the goal state is [false = true]), apply
    [ex_falso_quodlibet] to change the goal to [False].  This makes it
    easier to use assumptions of the form [~P] that may be available
    in the context -- in particular, assumptions of the form
    [x<>y]. *)

Theorem not_true_is_false : forall b : bool,
  b <> true -> b = false.
Proof.
  intros [] H.
  - (* b = true *)
    unfold not in H.
    apply ex_falso_quodlibet.
    apply H. reflexivity.
  - (* b = false *)
    reflexivity.
Qed.

(** Since reasoning with [ex_falso_quodlibet] is quite common, Coq
    provides a built-in tactic, [exfalso], for applying it. *)

Theorem not_true_is_false' : forall b : bool,
  b <> true -> b = false.
Proof.
  intros [] H.
  - (* b = false *)
    unfold not in H.
    exfalso.                (* <=== *)
    apply H. reflexivity.
  - (* b = true *) reflexivity.
Qed.

(* ================================================================= *)
(** ** Truth *)

(** Besides [False], Coq's standard library also defines [True], a
    proposition that is trivially true. To prove it, we use the
    predefined constant [I : True]: *)

(* NOTE: Since I used I for the binary exercise, I need to qualify it. *)
Lemma True_is_true : True.
Proof. apply Logic.I. Qed.

(** Unlike [False], which is used extensively, [True] is used quite
    rarely, since it is trivial (and therefore uninteresting) to prove
    as a goal, and it carries no useful information as a hypothesis.
    But it can be quite useful when defining complex [Prop]s using
    conditionals or as a parameter to higher-order [Prop]s.  We will
    see some examples such uses of [True] later on. *)

(* ================================================================= *)
(** ** Logical Equivalence *)

(** The handy "if and only if" connective, which asserts that two
    propositions have the same truth value, is just the conjunction of
    two implications. *)

Module MyIff.

Definition iff (P Q : Prop) := (P -> Q) /\ (Q -> P).

Notation "P <-> Q" := (iff P Q)
                      (at level 95, no associativity)
                      : type_scope.

End MyIff.

Theorem iff_sym : forall P Q : Prop,
  (P <-> Q) -> (Q <-> P).
Proof.
  (* WORKED IN CLASS *)
  intros P Q [HAB HBA].
  split.
  - (* -> *) apply HBA.
  - (* <- *) apply HAB.  Qed.

Lemma not_true_iff_false : forall b,
  b <> true <-> b = false.
Proof.
  (* WORKED IN CLASS *)
  intros b. split.
  - (* -> *) apply not_true_is_false.
  - (* <- *)
    intros H. rewrite H. intros H'. inversion H'.
Qed.

(** **** Exercise: 1 star, optional (iff_properties)  *)
(** Using the above proof that [<->] is symmetric ([iff_sym]) as
    a guide, prove that it is also reflexive and transitive. *)

Theorem iff_refl : forall P : Prop,
  P <-> P.
Proof.
  intros. unfold iff. split.
  - intros. assumption.
  - intros. assumption.
Qed.

Theorem iff_trans : forall P Q R : Prop,
  (P <-> Q) -> (Q <-> R) -> (P <-> R).
Proof.
  unfold iff. intros P Q R [] []. intros. split.
  - intros. apply (H1 (H H3)).
  - intros. apply (H0 (H2 H3)).
Qed.

(** [] *)

(** **** Exercise: 3 stars (or_distributes_over_and)  *)
Theorem or_distributes_over_and : forall P Q R : Prop,
  P \/ (Q /\ R) <-> (P \/ Q) /\ (P \/ R).
Proof.
  unfold iff. intros. split.
  - intros []. intros. split.
    + apply (or_intro _ _ H).
    + apply (or_intro _ _ H).
    + destruct H. split.
      * apply (or_intro_r _ _ H).
      * apply (or_intro_r _ _ H0).
  - intros [ [] [] ].
    + intros. left. assumption.
    + intros. left. assumption.
    + intros. left. assumption.
    + intros. right. apply (and_intro _ _ H H0).
Qed.

(** [] *)

(** Some of Coq's tactics treat [iff] statements specially, avoiding
    the need for some low-level proof-state manipulation.  In
    particular, [rewrite] and [reflexivity] can be used with [iff]
    statements, not just equalities.  To enable this behavior, we need
    to import a special Coq library that allows rewriting with other
    formulas besides equality: *)

Require Import Coq.Setoids.Setoid.

(** Here is a simple example demonstrating how these tactics work with
    [iff].  First, let's prove a couple of basic iff equivalences: *)

Lemma mult_0 : forall n m, n * m = 0 <-> n = 0 \/ m = 0.
Proof.
  split.
  - apply mult_eq_0.
  - apply or_example.
Qed.

Lemma or_assoc :
  forall P Q R : Prop, P \/ (Q \/ R) <-> (P \/ Q) \/ R.
Proof.
  intros P Q R. split.
  - intros [H | [H | H]].
    + left. left. apply H.
    + left. right. apply H.
    + right. apply H.
  - intros [[H | H] | H].
    + left. apply H.
    + right. left. apply H.
    + right. right. apply H.
Qed.

(** We can now use these facts with [rewrite] and [reflexivity] to
    give smooth proofs of statements involving equivalences.  Here is
    a ternary version of the previous [mult_0] result: *)

Lemma mult_0_3 :
  forall n m p, n * m * p = 0 <-> n = 0 \/ m = 0 \/ p = 0.
Proof.
  intros n m p.
  rewrite mult_0. rewrite mult_0. rewrite or_assoc.
  reflexivity.
Qed.

(** The [apply] tactic can also be used with [<->]. When given an
    equivalence as its argument, [apply] tries to guess which side of
    the equivalence to use. *)

Lemma apply_iff_example :
  forall n m : nat, n * m = 0 -> n = 0 \/ m = 0.
Proof.
  intros n m H. apply mult_0. apply H.
Qed.

(* ================================================================= *)
(** ** Existential Quantification *)

(** Another important logical connective is _existential
    quantification_.  To say that there is some [x] of type [T] such
    that some property [P] holds of [x], we write [exists x : T,
    P]. As with [forall], the type annotation [: T] can be omitted if
    Coq is able to infer from the context what the type of [x] should
    be.

    To prove a statement of the form [exists x, P], we must show that
    [P] holds for some specific choice of value for [x], known as the
    _witness_ of the existential.  This is done in two steps: First,
    we explicitly tell Coq which witness [t] we have in mind by
    invoking the tactic [exists t]; then we prove that [P] holds after
    all occurrences of [x] are replaced by [t].  Here is an example: *)

Lemma four_is_even : exists n : nat, 4 = n + n.
Proof.
  exists 2. reflexivity.
Qed.

(** Conversely, if we have an existential hypothesis [exists x, P] in
    the context, we can destruct it to obtain a witness [x] and a
    hypothesis stating that [P] holds of [x]. *)

Theorem exists_example_2 : forall n,
  (exists m, n = 4 + m) ->
  (exists o, n = 2 + o).
Proof.
  intros n [m Hm].
  exists (2 + m).
  apply Hm.  Qed.

(** **** Exercise: 1 star (dist_not_exists)  *)
(** Prove that "[P] holds for all [x]" implies "there is no [x] for
    which [P] does not hold." *)

Theorem dist_not_exists : forall (X:Type) (P : X -> Prop),
  (forall x, P x) -> ~ (exists x, ~ P x).
Proof.
  intros. unfold not. intros. destruct H0. apply (H0 (H x)).
Qed.

(** [] *)

(** **** Exercise: 2 stars (dist_exists_or)  *)
(** Prove that existential quantification distributes over
    disjunction. *)

Theorem dist_exists_or : forall (X:Type) (P Q : X -> Prop),
  (exists x, P x \/ Q x) <-> (exists x, P x) \/ (exists x, Q x).
Proof.
  intros. split.
  - intros. destruct H. destruct H.
    + left. exists x. assumption.
    + right. exists x. assumption.
  - intros. destruct H.
    + destruct H. exists x. left. assumption.
    + destruct H. exists x. right. assumption.
Qed.

(** [] *)

(* ################################################################# *)
(** * Programming with Propositions *)

(** The logical connectives that we have seen provide a rich
    vocabulary for defining complex propositions from simpler ones.
    To illustrate, let's look at how to express the claim that an
    element [x] occurs in a list [l].  Notice that this property has a
    simple recursive structure:

    - If [l] is the empty list, then [x] cannot occur on it, so the
      property "[x] appears in [l]" is simply false.

    - Otherwise, [l] has the form [x' :: l'].  In this case, [x]
      occurs in [l] if either it is equal to [x'] or it occurs in
      [l']. *)

(** We can translate this directly into a straightforward Coq
    function, [In].  (It can also be found in the Coq standard
    library.) *)
Require Import Coq.Lists.List.
Import ListNotations.

Open Scope list_scope.
Fixpoint In {A : Type} (x : A) (l : list A) : Prop :=
  match l with
  | [] => False
  | x' :: l' => x' = x \/ In x l'
  end.

(** When [In] is applied to a concrete list, it expands into a
    concrete sequence of nested conjunctions. *)

Example In_example_1 : In 4 [3; 4; 5].
Proof.
  simpl. right. left. reflexivity.
Qed.

Example In_example_2 :
  forall n, In n [2; 4] ->
  exists n', n = 2 * n'.
Proof.
  simpl.
  intros n [H | [H | []]].
  - exists 1. rewrite <- H. reflexivity.
  - exists 2. rewrite <- H. reflexivity.
Qed.

(** (Notice the use of the empty pattern to discharge the last case
    _en passant_.) *)

(** We can also prove more generic, higher-level lemmas about [In].
    Note, in the next, how [In] starts out applied to a variable and
    only gets expanded when we do case analysis on this variable: *)

Lemma In_map :
  forall (A B : Type) (f : A -> B) (l : list A) (x : A),
    In x l ->
    In (f x) (map f l).
Proof.
  intros A B f l x.
  induction l as [|x' l' IHl'].
  - (* l = nil, contradiction *)
    simpl. intros [].
  - (* l = x' :: l' *)
    simpl. intros [H | H].
    + rewrite H. left. reflexivity.
    + right. apply IHl'. apply H.
Qed.

(** This way of defining propositions, though convenient in some
    cases, also has some drawbacks.  In particular, it is subject to
    Coq's usual restrictions regarding the definition of recursive
    functions, e.g., the requirement that they be "obviously
    terminating."  In the next chapter, we will see how to define
    propositions _inductively_, a different technique with its own set
    of strengths and limitations. *)

(** **** Exercise: 2 stars (In_map_iff)  *)
Lemma In_map_iff :
  forall (A B : Type) (f : A -> B) (l : list A) (y : B),
    In y (map f l) <->
    exists x, f x = y /\ In x l.
Proof.
  induction l.
  - simpl. intros. unfold iff. split.
    + intros. contradict H.
    + intros. destruct H. destruct H. assumption.
  - simpl. intros. unfold iff. split.
    + intros. destruct H.
      * {exists x. split.
         - assumption.
         - left. reflexivity.
        }
      * { pose proof (IHl y). rewrite H0 in H. destruct H.
          - exists x0. split.
            + destruct H. assumption.
            + destruct H. right. assumption.
        }
    + intros. destruct H.
      * { destruct H. destruct H0.
          - left. rewrite <- H0 in H. assumption.
          - right. pose proof (IHl y). rewrite H1. exists x0. split.
            + assumption.
            + assumption.
          }
Qed.
(** [] *)

(** **** Exercise: 2 stars (in_app_iff)  *)
Lemma in_app_iff : forall A l l' (a:A),
  In a (l++l') <-> In a l \/ In a l'.
Proof.
  intros. split.
  - generalize dependent l'. induction l.
    + simpl. intros. right. assumption.
    + simpl. intros. destruct H.
      * left. left. assumption.
      * rewrite <- or_assoc. right. apply (IHl _ H).
  - intros. generalize dependent l'. induction l.
    + simpl. intros. destruct H.
      * contradiction H.
      * assumption.
    + simpl. intros. rewrite <- or_assoc in H. destruct H.
      * left. assumption.
      * right. apply (IHl _ H).
Qed.

(** [] *)

(** **** Exercise: 3 stars (All)  *)
(** Recall that functions returning propositions can be seen as
    _properties_ of their arguments. For instance, if [P] has type
    [nat -> Prop], then [P n] states that property [P] holds of [n].

    Drawing inspiration from [In], write a recursive function [All]
    stating that some property [P] holds of all elements of a list
    [l]. To make sure your definition is correct, prove the [All_In]
    lemma below.  (Of course, your definition should _not_ just
    restate the left-hand side of [All_In].) *)

Fixpoint All {T} (P : T -> Prop) (l : list T) : Prop :=
  match l with
    | [] => True
    | x :: xs => P x /\ All P xs
  end.

Lemma All_In :
  forall T (P : T -> Prop) (l : list T),
    (forall x, In x l -> P x) <->
    All P l.
Proof.
  intros. induction l.
  - simpl. unfold iff. split.
    + intros. apply Logic.I.
    + intros. contradict H0.
  - simpl. split.
    + intros. split.
      * apply H. left. reflexivity.
      * apply IHl. intros. apply H. right. assumption.
    + intros. destruct H. destruct H0.
      * rewrite <- H0. assumption.
      * rewrite <- IHl in H1. apply H1. assumption.
Qed.

(** [] *)

(** **** Exercise: 3 stars (combine_odd_even)  *)
(** Complete the definition of the [combine_odd_even] function below.
    It takes as arguments two properties of numbers, [Podd] and
    [Peven], and it should return a property [P] such that [P n] is
    equivalent to [Podd n] when [n] is odd and equivalent to [Peven n]
    otherwise. *)

Definition combine_odd_even (Podd Peven : nat -> Prop) : nat -> Prop :=
  fun (n:nat) => if (oddb n) then Podd n else Peven n.

(** To test your definition, prove the following facts: *)

Theorem combine_odd_even_intro :
  forall (Podd Peven : nat -> Prop) (n : nat),
    (oddb n = true -> Podd n) ->
    (oddb n = false -> Peven n) ->
    combine_odd_even Podd Peven n.
Proof.
  intros. simpl. unfold combine_odd_even. destruct (oddb n).
  - apply H. reflexivity.
  - apply H0. reflexivity.
Qed.

Theorem combine_odd_even_elim_odd :
  forall (Podd Peven : nat -> Prop) (n : nat),
    combine_odd_even Podd Peven n ->
    oddb n = true ->
    Podd n.
Proof.
  intros. unfold combine_odd_even in H. rewrite H0 in H. assumption.
Qed.

Theorem combine_odd_even_elim_even :
  forall (Podd Peven : nat -> Prop) (n : nat),
    combine_odd_even Podd Peven n ->
    oddb n = false ->
    Peven n.
Proof.
  intros. unfold combine_odd_even in H. rewrite H0 in H. assumption.
Qed.

(** [] *)

(* ################################################################# *)
(** * Applying Theorems to Arguments *)

(** One feature of Coq that distinguishes it from many other proof
    assistants is that it treats _proofs_ as first-class objects.

    There is a great deal to be said about this, but it is not
    necessary to understand it in detail in order to use Coq.  This
    section gives just a taste, while a deeper exploration can be
    found in the optional chapters [ProofObjects] and
    [IndPrinciples]. *)

(** We have seen that we can use the [Check] command to ask Coq to
    print the type of an expression.  We can also use [Check] to ask
    what theorem a particular identifier refers to. *)

Check plus_comm.
(* ===> forall n m : nat, n + m = m + n *)

(** Coq prints the _statement_ of the [plus_comm] theorem in the same
    way that it prints the _type_ of any term that we ask it to
    [Check].  Why?

    The reason is that the identifier [plus_comm] actually refers to a
    _proof object_ -- a data structure that represents a logical
    derivation establishing of the truth of the statement [forall n m
    : nat, n + m = m + n].  The type of this object _is_ the statement
    of the theorem that it is a proof of.

    Intuitively, this makes sense because the statement of a theorem
    tells us what we can use that theorem for, just as the type of a
    computational object tells us what we can do with that object --
    e.g., if we have a term of type [nat -> nat -> nat], we can give
    it two [nat]s as arguments and get a [nat] back.  Similarly, if we
    have an object of type [n = m -> n + n = m + m] and we provide it
    an "argument" of type [n = m], we can derive [n + n = m + m].

    Operationally, this analogy goes even further: by applying a
    theorem, as if it were a function, to hypotheses with matching
    types, we can specialize its result without having to resort to
    intermediate assertions.  For example, suppose we wanted to prove
    the following result: *)

Lemma plus_comm3 :
  forall n m p, n + (m + p) = (p + m) + n.

(** It appears at first sight that we ought to be able to prove this
    by rewriting with [plus_comm] twice to make the two sides match.
    The problem, however, is that the second [rewrite] will undo the
    effect of the first. *)

Proof.
  intros n m p.
  rewrite plus_comm.
  rewrite plus_comm.
  (* We are back where we started... *)

(** One simple way of fixing this problem, using only tools that we
    already know, is to use [assert] to derive a specialized version
    of [plus_comm] that can be used to rewrite exactly where we
    want. *)

  rewrite plus_comm.
  assert (H : m + p = p + m).
  { rewrite plus_comm. reflexivity. }
  rewrite H.
  reflexivity.
Qed.

(** A more elegant alternative is to apply [plus_comm] directly to the
    arguments we want to instantiate it with, in much the same way as
    we apply a polymorphic function to a type argument. *)

Lemma plus_comm3_take2 :
  forall n m p, n + (m + p) = (p + m) + n.
Proof.
  intros n m p.
  rewrite plus_comm.
  rewrite (plus_comm m).
  reflexivity.
Qed.

(** You can "use theorems as functions" in this way with almost all
    tactics that take a theorem name as an argument.  Note also that
    theorem application uses the same inference mechanisms as function
    application; thus, it is possible, for example, to supply
    wildcards as arguments to be inferred, or to declare some
    hypotheses to a theorem as implicit by default.  These features
    are illustrated in the proof below. *)

Example lemma_application_ex :
  forall {n : nat} {ns : list nat},
    In n (map (fun m => m * 0) ns) ->
    n = 0.
Proof.
  intros n ns H.
  destruct (proj1 _ _ (In_map_iff _ _ _ _ _) H)
           as [m [Hm _]].
  rewrite mult_0_r in Hm. rewrite <- Hm. reflexivity.
Qed.

(** We will see many more examples of the idioms from this section in
    later chapters. *)

(* ################################################################# *)
(** * Coq vs. Set Theory *)

(** Coq's logical core, the _Calculus of Inductive Constructions_,
    differs in some important ways from other formal systems that are
    used by mathematicians for writing down precise and rigorous
    proofs.  For example, in the most popular foundation for
    mainstream paper-and-pencil mathematics, Zermelo-Fraenkel Set
    Theory (ZFC), a mathematical object can potentially be a member of
    many different sets; a term in Coq's logic, on the other hand, is
    a member of at most one type.  This difference often leads to
    slightly different ways of capturing informal mathematical
    concepts, though these are by and large quite natural and easy to
    work with.  For example, instead of saying that a natural number
    [n] belongs to the set of even numbers, we would say in Coq that
    [ev n] holds, where [ev : nat -> Prop] is a property describing
    even numbers.

    However, there are some cases where translating standard
    mathematical reasoning into Coq can be either cumbersome or
    sometimes even impossible, unless we enrich the core logic with
    additional axioms.  We conclude this chapter with a brief
    discussion of some of the most significant differences between the
    two worlds. *)

(** ** Functional Extensionality

    The equality assertions that we have seen so far mostly have
    concerned elements of inductive types ([nat], [bool], etc.).  But
    since Coq's equality operator is polymorphic, these are not the
    only possibilities -- in particular, we can write propositions
    claiming that two _functions_ are equal to each other: *)

Example function_equality_ex : plus 3 = plus (pred 4).
Proof. reflexivity. Qed.

(** In common mathematical practice, two functions [f] and [g] are
    considered equal if they produce the same outputs:

    (forall x, f x = g x) -> f = g

    This is known as the principle of _functional extensionality_.

    Informally speaking, an "extensional property" is one that
    pertains to an object's observable behavior.  Thus, functional
    extensionality simply means that a function's identity is
    completely determined by what we can observe from it -- i.e., in
    Coq terms, the results we obtain after applying it.

    Functional extensionality is not part of Coq's basic axioms: the
    only way to show that two functions are equal is by
    simplification (as we did in the proof of [function_equality_ex]).
    But we can add it to Coq's core logic using the [Axiom]
    command. *)

Axiom functional_extensionality : forall {X Y: Type}
                                    {f g : X -> Y},
  (forall (x:X), f x = g x) -> f = g.

(** Using [Axiom] has the same effect as stating a theorem and
    skipping its proof using [Admitted], but it alerts the reader that
    this isn't just something we're going to come back and fill in
    later!

    We can now invoke functional extensionality in proofs: *)

Lemma plus_comm_ext : plus = fun n m => m + n.
Proof.
  apply functional_extensionality. intros n.
  apply functional_extensionality. intros m.
  apply plus_comm.
Qed.

(** Naturally, we must be careful when adding new axioms into Coq's
    logic, as they may render it inconsistent -- that is, it may
    become possible to prove every proposition, including [False]!
    Unfortunately, there is no simple way of telling whether an axiom
    is safe: hard work is generally required to establish the
    consistency of any particular combination of axioms.  Fortunately,
    it is known that adding functional extensionality, in particular,
    _is_ consistent.

    Note that it is possible to check whether a particular proof
    relies on any additional axioms, using the [Print Assumptions]
    command. For instance, if we run it on [plus_comm_ext], we see
    that it uses [functional_extensionality]: *)

Print Assumptions plus_comm_ext.
(* ===>
     Axioms:
     functional_extensionality :
         forall (X Y : Type) (f g : X -> Y),
                (forall x : X, f x = g x) -> f = g *)

(** **** Exercise: 5 stars (tr_rev)  *)
(** One problem with the definition of the list-reversing function
    [rev] that we have is that it performs a call to [app] on each
    step; running [app] takes time asymptotically linear in the size
    of the list, which means that [rev] has quadratic running time.
    We can improve this with the following definition: *)

Fixpoint rev_append {X} (l1 l2 : list X) : list X :=
  match l1 with
  | [] => l2
  | x :: l1' => rev_append l1' (x :: l2)
  end.

Definition tr_rev {X} (l : list X) : list X :=
  rev_append l [].

(** This version is said to be _tail-recursive_, because the recursive
    call to the function is the last operation that needs to be
    performed (i.e., we don't have to execute [++] after the recursive
    call); a decent compiler will generate very efficient code in this
    case.  Prove that both definitions are indeed equivalent. *)

Lemma rev_append_app : forall {X : Type} (l1 : list X) (l2 : list X),
    rev_append l1 l2 = rev_append l1 [] ++ l2.
Proof.
  induction l1.
  - reflexivity.
  - simpl. intros. rewrite (IHl1 (x :: l2)). rewrite (IHl1 [x]).
    rewrite <- app_assoc. reflexivity.
Qed.

Lemma tr_rev_correct : forall X, @tr_rev X = @rev X.
  intros. apply functional_extensionality.
  intros l. induction l.
  - reflexivity.
  - simpl. rewrite <- IHl. unfold tr_rev. simpl. rewrite rev_append_app.
    reflexivity.
Qed.

(** [] *)

(* ================================================================= *)
(** ** Propositions and Booleans *)

(** We've seen that Coq has two different ways of encoding logical
    facts: with _booleans_ (of type [bool]), and with
    _propositions_ (of type [Prop]). For instance, to claim that a
    number [n] is even, we can say either (1) that [evenb n] returns
    [true] or (2) that there exists some [k] such that [n = double k].
    Indeed, these two notions of evenness are equivalent, as can
    easily be shown with a couple of auxiliary lemmas (one of which is
    left as an exercise).

    We often say that the boolean [evenb n] _reflects_ the proposition
    [exists k, n = double k].  *)

Theorem evenb_double : forall k, evenb (double k) = true.
Proof.
  intros k. induction k as [|k' IHk'].
  - reflexivity.
  - simpl. apply IHk'.
Qed.

(** **** Exercise: 3 stars (evenb_double_conv)  *)
Theorem evenb_double_conv : forall n,
  exists k, n = if evenb n then double k
                else S (double k).
Proof.
  intros. induction n.
  - simpl. exists 0. reflexivity.
  - destruct IHn. destruct (evenb n).
    + rewrite evenb_S. rewrite H. exists x. rewrite evenb_double. simpl. reflexivity.
    + rewrite H. simpl. rewrite evenb_double. exists (S x). reflexivity.
Qed.

(** [] *)

Theorem even_bool_prop : forall n,
  evenb n = true <-> exists k, n = double k.
Proof.
  intros n. split.
  - intros H. destruct (evenb_double_conv n) as [k Hk].
    rewrite Hk. rewrite H. exists k. reflexivity.
  - intros [k Hk]. rewrite Hk. apply evenb_double.
Qed.

(** Similarly, to state that two numbers [n] and [m] are equal, we can
    say either (1) that [beq_nat n m] returns [true] or (2) that [n =
    m].  These two notions are equivalent. *)

Theorem beq_nat_true_iff : forall n1 n2 : nat,
  beq_nat n1 n2 = true <-> n1 = n2.
Proof.
  intros n1 n2. split.
  - apply beq_nat_true.
  - intros H. rewrite H. rewrite <- beq_nat_refl. reflexivity.
Qed.

(** However, while the boolean and propositional formulations of a
    claim are equivalent from a purely logical perspective, we have
    also seen that they need not be equivalent _operationally_.
    Equality provides an extreme example: knowing that [beq_nat n m =
    true] is generally of little help in the middle of a proof
    involving [n] and [m]; however, if we convert the statement to the
    equivalent form [n = m], we can rewrite with it.

    The case of even numbers is also interesting.  Recall that, when
    proving the backwards direction of
    [even_bool_prop] ([evenb_double], going from the propositional to
    the boolean claim), we used a simple induction on [k]).  On the
    other hand, the converse (the [evenb_double_conv] exercise)
    required a clever generalization, since we can't directly prove
    [(exists k, n = double k) -> evenb n = true].

    For these examples, the propositional claims were more useful than
    their boolean counterparts, but this is not always the case.  For
    instance, we cannot test whether a general proposition is true or
    not in a function definition; as a consequence, the following code
    fragment is rejected: *)

Fail Definition is_even_prime n :=
  if n = 2 then true
  else false.

(** Coq complains that [n = 2] has type [Prop], while it expects an
    elements of [bool] (or some other inductive type with two
    elements).  The reason for this error message has to do with the
    _computational_ nature of Coq's core language, which is designed
    so that every function that it can express is computable and
    total.  One reason for this is to allow the extraction of
    executable programs from Coq developments.  As a consequence,
    [Prop] in Coq does _not_ have a universal case analysis operation
    telling whether any given proposition is true or false, since such
    an operation would allow us to write non-computable functions.

    Although general non-computable properties cannot be phrased as
    boolean computations, it is worth noting that even many
    _computable_ properties are easier to express using [Prop] than
    [bool], since recursive function definitions are subject to
    significant restrictions in Coq.  For instance, the next chapter
    shows how to define the property that a regular expression matches
    a given string using [Prop].  Doing the same with [bool] would
    amount to writing a regular expression matcher, which would be
    more complicated, harder to understand, and harder to reason
    about.

    Conversely, an important side benefit of stating facts using
    booleans is enabling some proof automation through computation
    with Coq terms, a technique known as _proof by
    reflection_. Consider the following statement: *)

Example even_1000 : exists k, 1000 = double k.

(** The most direct proof of this fact is to give the value of [k]
    explicitly. *)

 Proof. exists 500. reflexivity. Qed.

(** On the other hand, the proof of the corresponding boolean
    statement is even simpler: *)

Example even_1000' : evenb 1000 = true.
Proof. reflexivity. Qed.

(** What is interesting is that, since the two notions are equivalent,
    we can use the boolean formulation to prove the other one without
    mentioning 500 explicitly: *)

Example even_1000'' : exists k, 1000 = double k.
Proof. apply even_bool_prop. reflexivity. Qed.

(** Although we haven't gained much in terms of proof size in this
    case, larger proofs can often be made considerably simpler by the
    use of reflection.  As an extreme example, the Coq proof of the
    famous _4-color theorem_ uses reflection to reduce the analysis of
    hundreds of different cases to a boolean computation.  We won't
    cover reflection in great detail, but it serves as a good example
    showing the complementary strengths of booleans and general
    propositions. *)

(** **** Exercise: 2 stars (logical_connectives)  *)
(** The following lemmas relate the propositional connectives studied
    in this chapter to the corresponding boolean operations. *)

Lemma andb_true_iff : forall b1 b2:bool,
  b1 && b2 = true <-> b1 = true /\ b2 = true.
Proof.
  intros. split.
  - intros. split.
    + rewrite andb_commutative in H. apply (andb_true_elim2 _ _ H).
    + apply (andb_true_elim2 _ _ H).
  - intros. destruct H. rewrite H. rewrite H0. reflexivity.
Qed.

Lemma orb_true_iff : forall b1 b2,
  b1 || b2 = true <-> b1 = true \/ b2 = true.
Proof.
  intros. split.
  - intros. destruct b1.
    + left. reflexivity.
    + destruct b2.
      * right. reflexivity.
      * inversion H.
  - intros. destruct H.
    + rewrite H. reflexivity.
    + rewrite H. destruct b1.
      * reflexivity.
      * reflexivity.
Qed.

(** [] *)

(** **** Exercise: 1 star (beq_nat_false_iff)  *)
(** The following theorem is an alternate "negative" formulation of
    [beq_nat_true_iff] that is more convenient in certain
    situations (we'll see examples in later chapters). *)

Theorem beq_nat_false_iff : forall x y : nat,
  beq_nat x y = false <-> x <> y.
Proof.
  intros. split.
  - intros. unfold not. intros. rewrite H0 in H.
    assert (forall n, beq_nat n n = true).
    + induction n.
      * reflexivity.
      * simpl. rewrite IHn. reflexivity.
    + rewrite (H1 y) in H. inversion H.
  - unfold not. intros. destruct (beq_nat x y) eqn:H0.
    + pose proof (beq_nat_true x y H0). contradiction (H H1).
    + reflexivity.
Qed.

(** [] *)

(** **** Exercise: 3 stars (beq_list)  *)
(** Given a boolean operator [beq] for testing equality of elements of
    some type [A], we can define a function [beq_list beq] for testing
    equality of lists with elements in [A].  Complete the definition
    of the [beq_list] function below.  To make sure that your
    definition is correct, prove the lemma [beq_list_true_iff]. *)

Fixpoint beq_list {A} (beq : A -> A -> bool)
         (l1 l2 : list A) : bool :=
  match l1, l2 with
  | [], [] => true
  | x :: xs, y :: ys => beq x y && beq_list beq xs ys
  | _ , _ => false
  end.

Lemma beq_list_true_iff :
  forall A (beq : A -> A -> bool),
    (forall a1 a2, beq a1 a2 = true <-> a1 = a2) ->
    forall l1 l2, beq_list beq l1 l2 = true <-> l1 = l2.
Proof.
  intros A beq H l1. unfold iff. induction l1.
  - split.
    + intros. destruct l2.
      * reflexivity.
      * inversion H0.
    + intros. destruct l2.
      * reflexivity.
      * inversion H0.
  - split.
    + intros. destruct l2.
      * inversion H0.
        * simpl in H0.
          rewrite andb_commutative in H0.
          pose proof (andb_true_elim2 _ _ H0).
          rewrite (H x a) in H1. rewrite <- H1. apply f_equal.
          rewrite andb_commutative in H0.
          pose proof (andb_true_elim2 _ _ H0).
          pose proof (IHl1 l2) as H3. destruct H3.
          apply (H3 H2).
    + intros. rewrite <- H0.
      pose proof IHl1 (l1) as H1. destruct H1. simpl. rewrite H2.
      * {assert (beq x x = true).
         - rewrite H. reflexivity.
         - rewrite H3. reflexivity.
        }
      * reflexivity.
Qed.

(** [] *)

(** **** Exercise: 2 stars, recommended (All_forallb)  *)
(** Recall the function [forallb], from the exercise
    [forall_exists_challenge] in chapter [Tactics]: *)

Fixpoint forallb {X : Type} (test : X -> bool) (l : list X) : bool :=
  match l with
  | [] => true
  | x :: l' => andb (test x) (forallb test l')
  end.

(** Prove the theorem below, which relates [forallb] to the [All]
    property of the above exercise. *)

Theorem forallb_true_iff : forall X test (l : list X),
   forallb test l = true <-> All (fun x => test x = true) l.
Proof.
  intros. induction l.
  - split.
    + intros. reflexivity.
    + intros. reflexivity.
  - split.
    + simpl. intros. split.
      * rewrite andb_commutative in H.
        apply (andb_true_elim2 _ _ H).
      * pose proof (andb_true_elim2 _ _ H).
        rewrite IHl in H0. apply H0.
    + simpl. intros. destruct H. rewrite H. rewrite IHl. apply H0.
Qed.

(** Are there any important properties of the function [forallb] which
    are not captured by your specification? *)

(* FILL IN HERE *)
(** [] *)

(* ================================================================= *)
(** ** Classical vs. Constructive Logic *)

(** We have seen that it is not possible to test whether or not a
    proposition [P] holds while defining a Coq function.  You may be
    surprised to learn that a similar restriction applies to _proofs_!
    In other words, the following intuitive reasoning principle is not
    derivable in Coq: *)

Definition excluded_middle := forall P : Prop,
  P \/ ~ P.

(** To understand operationally why this is the case, recall that, to
    prove a statement of the form [P \/ Q], we use the [left] and
    [right] tactics, which effectively require knowing which side of
    the disjunction holds.  However, the universally quantified [P] in
    [excluded_middle] is an _arbitrary_ proposition, which we know
    nothing about.  We don't have enough information to choose which
    of [left] or [right] to apply, just as Coq doesn't have enough
    information to mechanically decide whether [P] holds or not inside
    a function.  On the other hand, if we happen to know that [P] is
    reflected in some boolean term [b], then knowing whether it holds
    or not is trivial: we just have to check the value of [b].  This
    leads to the following theorem: *)

Theorem restricted_excluded_middle : forall P b,
  (P <-> b = true) -> P \/ ~ P.
Proof.
  intros P [] H.
  - left. rewrite H. reflexivity.
  - right. rewrite H. intros contra. inversion contra.
Qed.

(** In particular, the excluded middle is valid for equations [n = m],
    between natural numbers [n] and [m].

    You may find it strange that the general excluded middle is not
    available by default in Coq; after all, any given claim must be
    either true or false.  Nonetheless, there is an advantage in not
    assuming the excluded middle: statements in Coq can make stronger
    claims than the analogous statements in standard mathematics.
    Notably, if there is a Coq proof of [exists x, P x], it is
    possible to explicitly exhibit a value of [x] for which we can
    prove [P x] -- in other words, every proof of existence is
    necessarily _constructive_.  Because of this, logics like Coq's,
    which do not assume the excluded middle, are referred to as
    _constructive logics_.  More conventional logical systems such as
    ZFC, in which the excluded middle does hold for arbitrary
    propositions, are referred to as _classical_.

    The following example illustrates why assuming the excluded middle
    may lead to non-constructive proofs: *)

(** _Claim_: There exist irrational numbers [a] and [b] such that [a ^
    b] is rational.

    _Proof_: It is not difficult to show that [sqrt 2] is irrational.
    If [sqrt 2 ^ sqrt 2] is rational, it suffices to take [a = b =
    sqrt 2] and we are done.  Otherwise, [sqrt 2 ^ sqrt 2] is
    irrational.  In this case, we can take [a = sqrt 2 ^ sqrt 2] and
    [b = sqrt 2], since [a ^ b = sqrt 2 ^ (sqrt 2 * sqrt 2) = sqrt 2 ^
    2 = 2].  []

    Do you see what happened here?  We used the excluded middle to
    consider separately the cases where [sqrt 2 ^ sqrt 2] is rational
    and where it is not, without knowing which one actually holds!
    Because of that, we wind up knowing that such [a] and [b] exist
    but we cannot determine what their actual values are (at least,
    using this line of argument).

    As useful as constructive logic is, it does have its limitations:
    There are many statements that can easily be proven in classical
    logic but that have much more complicated constructive proofs, and
    there are some that are known to have no constructive proof at
    all!  Fortunately, like functional extensionality, the excluded
    middle is known to be compatible with Coq's logic, allowing us to
    add it safely as an axiom.  However, we will not need to do so in
    this book: the results that we cover can be developed entirely
    within constructive logic at negligible extra cost.

    It takes some practice to understand which proof techniques must
    be avoided in constructive reasoning, but arguments by
    contradiction, in particular, are infamous for leading to
    non-constructive proofs.  Here's a typical example: suppose that
    we want to show that there exists [x] with some property [P],
    i.e., such that [P x].  We start by assuming that our conclusion
    is false; that is, [~ exists x, P x]. From this premise, it is not
    hard to derive [forall x, ~ P x].  If we manage to show that this
    intermediate fact results in a contradiction, we arrive at an
    existence proof without ever exhibiting a value of [x] for which
    [P x] holds!

    The technical flaw here, from a constructive standpoint, is that
    we claimed to prove [exists x, P x] using a proof of [~ ~ exists
    x, P x]. However, allowing ourselves to remove double negations
    from arbitrary statements is equivalent to assuming the excluded
    middle, as shown in one of the exercises below.  Thus, this line
    of reasoning cannot be encoded in Coq without assuming additional
    axioms. *)

(** **** Exercise: 3 stars (excluded_middle_irrefutable)  *)
(** The consistency of Coq with the general excluded middle axiom
    requires complicated reasoning that cannot be carried out within
    Coq itself.  However, the following theorem implies that it is
    always safe to assume a decidability axiom (i.e., an instance of
    excluded middle) for any _particular_ Prop [P].  Why? Because we
    cannot prove the negation of such an axiom; if we could, we would
    have both [~ (P \/ ~P)] and [~ ~ (P \/ ~P)], a contradiction. *)

Theorem excluded_middle_irrefutable:  forall (P:Prop),
  ~ ~ (P \/ ~ P).
Proof.
  unfold not. intros. apply H. right. intros. apply H.
  apply (or_intro _ _ H0).
Qed.

(** [] *)

(** **** Exercise: 3 stars, optional (not_exists_dist)  *)
(** It is a theorem of classical logic that the following two
    assertions are equivalent:

    ~ (exists x, ~ P x)
    forall x, P x

    The [dist_not_exists] theorem above proves one side of this
    equivalence. Interestingly, the other direction cannot be proved
    in constructive logic. Your job is to show that it is implied by
    the excluded middle. *)

Theorem not_exists_dist :
  excluded_middle ->
  forall (X:Type) (P : X -> Prop),
    ~ (exists x, ~ P x) -> (forall x, P x).
Proof.
  intros. unfold excluded_middle in H. unfold not in H0. destruct (H (P x)).
  - apply H1.
  - unfold not in H1. exfalso. apply H0. exists x. apply H1.
Qed.

(** [] *)

(** **** Exercise: 5 stars, advanced, optional (classical_axioms)  *)
(** For those who like a challenge, here is an exercise taken from the
    Coq'Art book by Bertot and Casteran (p. 123).  Each of the
    following four statements, together with [excluded_middle], can be
    considered as characterizing classical logic.  We can't prove any
    of them in Coq, but we can consistently add any one of them as an
    axiom if we wish to work in classical logic.

    Prove that all five propositions (these four plus
    [excluded_middle]) are equivalent. *)

Definition peirce := forall P Q: Prop,
  ((P->Q)->P)->P.

Definition double_negation_elimination := forall P:Prop,
  ~~P -> P.

Definition de_morgan_not_and_not := forall P Q:Prop,
  ~(~P /\ ~Q) -> P\/Q.

Definition implies_to_or := forall P Q:Prop,
  (P->Q) -> (~P\/Q).

Theorem em_peirce : excluded_middle -> peirce.
Proof.
  unfold excluded_middle. unfold peirce. intros. unfold not in H.
  destruct (H P).
  - assumption.
  - apply H0. intros. exfalso. apply H1. apply H2.
Qed.

Theorem peirce_dne : peirce -> double_negation_elimination.
Proof.
  unfold peirce. unfold double_negation_elimination. intros.
  unfold not in H0. apply (H P False). intros. exfalso. apply H0.
  intros. apply (H1 H2).
Qed.

Theorem dne_de_morgan : double_negation_elimination -> de_morgan_not_and_not.
Proof.
  unfold double_negation_elimination. unfold de_morgan_not_and_not. unfold not.
  intros. apply H. intros. apply H0. split.
  + intros. apply H1. left. assumption.
  + intros. apply H1. right. assumption.
Qed.

Theorem de_morgan_implies_to_or : de_morgan_not_and_not -> implies_to_or.
Proof.
  unfold de_morgan_not_and_not. unfold implies_to_or. unfold not.
  intros. apply H. intros. destruct H1. apply H1. intros. apply H2.
  apply (H0 H3).
Qed.

Theorem implies_to_or_excluded_middle : implies_to_or -> excluded_middle.
Proof.
  unfold implies_to_or. unfold excluded_middle. unfold not.
  intros. apply or_commut. apply H. intros. assumption.
Qed.

(** [] *)

(** $Date: 2015-08-11 12:03:04 -0400 (Tue, 11 Aug 2015) $ *)


(* La librairie strandard est décrite ici

   https://coq.inria.fr/distrib/current/stdlib/

*)

Require Import List. (* Voir https://coq.inria.fr/distrib/current/stdlib/Coq.Lists.List.html *)

Reserved Notation "x '~p' y" (at level 70, no associativity).

Section list_perm.

  Variable X : Type.

  Inductive perm : list X -> list X -> Prop :=

    | perm_nil   :                    nil ~p nil

    | perm_cons  : forall x l1 l2,     l1 ~p l2
                               ->   x::l1 ~p x::l2

    | perm_swap  : forall x y l,  x::y::l ~p y::x::l

    | perm_trans : forall l1 l2 l3,    l1 ~p l2
                               ->      l2 ~p l3
                               ->      l1 ~p l3

  where "x '~p' y " := (perm x y).

  Fact perm_refl l : l ~p l. (* Cours *)
  Proof.
   induction l as [
                  |list head IH]. (*Division en deux subgoals*)
   apply perm_nil. (*Afin d'enlever le premier subgoal*)
   apply perm_cons. (* Afin d'enlever le second subgoal*)
   apply IH. (*Appliquer IH pour enlever le dernier subgoal*)
  Qed.

  Fact perm_length l1 l2 : l1 ~p l2 -> length l1 = length l2.
  Proof.
    intros H. (*Introduire hypothèse qui dit que l1 permute avec l2*)
    induction H as [
                   | x l1 l2 H1 IH1 (*Si on ajoute x à l1 et à l2 et que length l1 = length l2
                   alors elles font toujours la même taille*)
                   | x y l (*Ajouter deux éléments pas dans le même ordre
                   pour montrer que l'ordre n'influe pas*)
                   | l1 l2 l3 H1 IH1 H2 IH2
                   ].
    apply refl_equal. (* Reflexivity :S'applique sur le but length nil = length nil*)
    simpl. (* Permet d'enlever les x:: et de travailler directement
    sur les listes *)
    f_equal. (* Permet d'éliminer la fonction S des deux côtés*)
    apply IH1. (* Appliquer IH1 *)
    simpl. (* Enleve les x et y des sous buts *)
    apply refl_equal. (* Reflexivity *)
    transitivity (length l2). (* Transitivité sur l2 *)
    apply IH1.
    apply IH2.
   Qed.

  Fact perm_sym l1 l2 : l1 ~p l2 -> l2 ~p l1.
  Proof.
   intros H.
   induction H. (* Création des 4 cas possibles *)
   apply perm_nil.
   apply perm_cons.
   assumption. (* Applique IHperm *)
   apply perm_swap.
   apply perm_trans with l2. (* Pour pouvoir utiliser IHperm1 et IHperm2 *)
   apply IHperm2.
   apply IHperm1.
  Qed.

  Fact perm_middle x l r : x::l++r ~p l++x::r.
  Proof.
    induction l as [ | y list IHl ]. (* 1 sous but avec le cas de la liste vide et
    un sous but avec le swap *)
    simpl. (* Enlever les nil *)
    apply perm_refl. (* Permet d'enlever un sous but *)
    simpl. (*Permet d'enlever les parenthèses inutiles *)
    apply perm_trans with (1 := perm_swap _ _ _).
    apply perm_cons.
    apply IHl.
  Qed.

  Let perm_app_left l r1 r2 : r1 ~p r2 -> l++r1 ~p l++r2.
  Proof.
    intros H.
    induction l. (*Sous but avec le cas de la liste vide
    et un sous but avec le swap*)
    simpl. (* Enlever les nil *)
    apply H.
    simpl. (* Enlever parenthèses*)
    apply perm_cons.
    apply IHl.
  Qed.

  Fact perm_app l1 l2 r1 r2 : l1 ~p l2 -> r1 ~p r2 -> l1++r1 ~p l2++r2.
  Proof.
   intros H.
   revert H r1 r2. (* Rentre les hypothèses dans le subgoal *)
   intros H.
   induction H as [
                   | x l1 l2 H1 IH1
                   | x y l
                   | l1 l2 l3 H1 IH1 H2 IH2
                   ].
    simpl . (* Enlever les nil *)
    intros H3 H4 H5.
    apply H5.  (*Enlever le premier subgoal*)
    intros H3 H4 H5.
    simpl.
    apply perm_cons. (*Permet d'enlever les x*)
    apply IH1.
    apply H5. (* Enlever le second subgoal *)
    intros H3 H4 H5.
    simpl.
    apply perm_trans with (1 := perm_swap _ _ _). (*Permet de rendre le memvre de gauche
    identique à celui de droite*)
    apply perm_cons. (*enlève y*)
    apply perm_cons. (*Enlève x*)
    apply perm_app_left.
    apply H5. (* Enlève le troisième sous but *)
    (*subgoal 4 *)
    intros H3 H4 H5.
    apply perm_trans with (l2 ++ H4). (* Faire le lien entre l1 et l3 *)
    apply IH1.
    apply H5.
    apply perm_trans with (l3++H3).
    apply IH2.
    apply perm_sym. (* Pour appliquer H5 *)
    apply H5.
    apply perm_app_left.
    apply H5. (* Résolution du dernier sous but *)
   Qed.

  (* incl est défini dans la librairie standard, fichier List.v *)

  Print incl.

  Fact perm_incl l m : l ~p m -> incl l m.
  Proof.
    intros H.
    induction H as [
                   | x l1 l2 H1 IH1
                   | x y l
                   | l1 l2 l3 H1 IH1 H2 IH2
                   ].
    intros Hyp1 Hyp2 .
    apply Hyp2.
    intros Hyp1 Hyp2 .
    destruct Hyp2. (* *)
    left. (* Afin de récupérer x = Hyp1 *)
    apply H.
    right. (* Afin de récupérer l2 = Hyp1 *)
    apply IH1. (* Pour pouvoir utiliser H ensuite *)
    apply H.
    intros ? ?.
    revert H.
    simpl. (* Pour faire fonctionner le tauto *)
    tauto. (* Pas arrivé à trouver autrement *)
    revert IH2.
    revert IH1.
    apply incl_tran. (* Transitivité sur les listes *)
  Qed.

End list_perm.

Infix "~p" := (perm _) (at level 70, no associativity).


Add LoadPath "/Users/Henry/Documents/Drive/Coq/Math/sorting".
Load natlist.

Fixpoint eqb (n m : nat) :=
  match n,m with
  | O,O => true
  | O,_ => false
  | _,O => false
  | S n,S m => eqb n m
  end.

Fixpoint leb (n m : nat) :=
  match n,m with
  | O,_ => true
  | _,O => false
  | S n,S m => leb n m
  end.

Definition can_head (n : nat) (l : natlist) :=
  match l with
  | []   => true
  | m::_ => leb n m
  end.

Inductive sorted : natlist -> Prop :=
  | sorted_nil  : sorted []
  | sorted_cons : forall (n : nat) (l : natlist),
      sorted l -> can_head n l = true -> sorted (cons n l).

Example sorted_singleton : forall n, sorted (cons n nil).
Proof.
  intros.
  refine (sorted_cons _ _ _ _ ).
  apply sorted_nil.
  exact (eq_refl : can_head n [] = true).
Qed.

Ltac red_sorted_cons := (refine (sorted_cons _ _ _ _)).

(* Example example1 : sorted (cons 1 (cons 2 (cons 3 nil))). *)
Example sorted_example1 : sorted [1;2;3].
Proof.
  red_sorted_cons. red_sorted_cons. red_sorted_cons.
  apply sorted_nil.
  apply eq_refl. apply eq_refl. apply eq_refl.
Qed.

Example sorted_example2 : ~ sorted [2;1].
Proof.
  unfold not.
  intros.
  inversion H.
  simpl in H3.
  discriminate.
Qed.

Example sorted_example3 : forall n, sorted [n;S n].
Proof.
  intros. red_sorted_cons. red_sorted_cons.
  apply sorted_nil.
  simpl. reflexivity.
  simpl. cut (leb n (S n) = true). intros. exact H.
  induction n.
    simpl. reflexivity.
    simpl. apply IHn.
Qed.